CREATE OR REPLACE PACKAGE BODY sweeps
/*******************************************************************************   
sweeps   
   
<pre>   
Artisan      Date      Comments   
============ ========= ========================================================   
Lobo-Blanco  2009Feb01 RECORDS-1179 Created for sweep feature.  
bcoulam      2009Feb27 RECORDS-1179 Rewrite.  
bcoulam      2009Mar20 RECORDS-1029 Adding sweep tally code.  
bcoulam      2009Jun12 RECORDS-2138 Rewrite.  
bcoulam      2009Aug03 RECORDS-3702 Mucho modifications. 
bcoulam      2010Dec13 RECORDS-4590 Removed rspbl_org_sens_flg column.  
</pre>   
  
                   Copyright 2009 Intellectual Reserve, Inc.                    
*******************************************************************************/   
AS   
   
--------------------------------------------------------------------------------  
--                 PACKAGE CONSTANTS, VARIABLES, TYPES, EXCEPTIONS  
--------------------------------------------------------------------------------  
gc_start CONSTANT VARCHAR2(10) := 'START';  
gc_update CONSTANT VARCHAR2(10) := 'UPDATE';  
gc_end   CONSTANT VARCHAR2(10) := 'END';  
  
--------------------------------------------------------------------------------  
--                        PRIVATE FUNCTIONS AND PROCEDURES  
--------------------------------------------------------------------------------  
  
/**-----------------------------------------------------------------------------  
load_sweep_set  
 Fills the temporary table with IDs of those missionaries that need to be picked  
 up for either sweep or tally purposes. Since the tally needs to count prefield  
 missionaries, we end up including a number of prefield missionaries who probably  
 aren't needed in the sweep, but that is how legacy did it, which we were told  
 to replicate.  
------------------------------------------------------------------------------*/  
PROCEDURE load_sweep_set(i_sweep_dt IN DATE)  
IS  
BEGIN  
   DELETE FROM fin_sweep_msny_gt;  
     
   INSERT INTO fin_sweep_msny_gt  
      SELECT m.id  
            ,m.enamem_id  
            ,m.legacy_miss_id  
            ,m.mrn  
            ,mn.first_nm||DECODE(em.misstyp_id,3,' and '||mnw.first_nm,NULL)
            ,mn.last_nm
            ,DECODE(em.misstyp_id,3,NULL,mn.middle_nm)
            ,em.term_months pmts_to_send  
            ,SUM(DECODE(fph.dr_cr_code, 'DR', 1, 'CR', -1)) pmts_sent  
        FROM enabled_members em
        JOIN missionaries m ON m.legacy_miss_id = em.legacy_miss_id -- only joins husband for couples
        JOIN current_perm_assignment_vw cpa ON cpa.call_emem_id = em.enamem_id AND cpa.early_release_flag <> 'Y' -- RMH Records-4102 Filter out early released missionaries
        JOIN msny_name mn ON mn.legacy_miss_id = em.legacy_miss_id AND mn.name_type_id = 10 -- Candidate name 
        LEFT JOIN msny_name mnw ON mnw.legacy_miss_id = em.legacy_spouse_miss_id AND mnw.name_type_id = 10 -- Wife's Candidate name 
        LEFT JOIN fin_pay_hist fph ON fph.legacy_miss_id = m.legacy_miss_id
       WHERE em.procstat_id IN (11, 12, 13, 14)  
         AND em.enamem_release_date >= i_sweep_dt  
       GROUP BY m.id  
               ,m.enamem_id  
               ,m.legacy_miss_id  
               ,m.mrn  
               ,mn.first_nm||DECODE(em.misstyp_id,3,' and '||mnw.first_nm,NULL)
               ,mn.last_nm
               ,DECODE(em.misstyp_id,3,NULL,mn.middle_nm)
               ,em.term_months;  
  
   -- This was the new filter, rejected in favor of the Legacy filter used above.  
   -- The legacy filter currently (July 2009) includes about 5000+ more missionaries  
   -- than the new filter would have.  
   --    (  
   --     -- 12=Started, 13=Arrived, 43=On leave from MTC, 44= On leave from Field  
   --     em.procstat_id IN (12, 13, 43, 44)  
   --     OR  
   --      -- 11=Assigned and eff dt of sweep is same as month of mtc start  
   --     (em.procstat_id = 11 AND TRUNC(em.enamem_mission_start_date, 'Month') = i_sweep_dt)  
   --     OR  
   --      -- 14=Released, but date of sweep is same as month of release  
   --     (em.procstat_id = 14 AND TRUNC(em.enamem_release_date, 'Month') = i_sweep_dt)  
   --    )    
       
END load_sweep_set;  
  
/**-----------------------------------------------------------------------------  
sweep_finalized  
 Returns TRUE if the last run for the given sweep month was already Finalized,  
 else it returns FALSE. This allows users to attempt the sweep multiple times,  
 but only one capped off as FINALIZED will prevent further sweep runs for the  
 same month.  
------------------------------------------------------------------------------*/  
FUNCTION sweep_finalized(i_sweep_dt IN DATE)  
  RETURN BOOLEAN  
IS  
   l_finalized fin_proc_hist.status_cd%TYPE;  
BEGIN  
   -- check for the existence of the Finalized tag.  
   SELECT status_cd  
     INTO l_finalized  
     FROM fin_proc_hist  
    WHERE proc_hist_id = (SELECT MAX(proc_hist_id)  
                            FROM fin_proc_hist  
                           WHERE proc_nm = sweep_process_nm  
                             AND proc_dt = i_sweep_dt);  
   IF (INSTR(l_finalized, 'FINALIZED') > 0) THEN  
      RETURN TRUE;  
   ELSE  
      RETURN FALSE;  
   END IF;  
                                  
EXCEPTION  
   WHEN NO_DATA_FOUND THEN  
      RETURN FALSE;                               
END sweep_finalized;  
  
/**-----------------------------------------------------------------------------  
tag_proc_hist  
 Used to note the start, progress and finish of a sweep run. If there are unexpected  
 errors the end_dt will be empty and the process text field will still indicate  
 it has only started.  
  
%param i_sweep_dt Date of the sweeps being tracked in fin_proc_hist.  
%param i_op Whether or not this tagging operation is due to a process starting   
            (use global constant gc_INSERT), noting a recent state change (use  
            constant gc_UPDATE), or a process ending (use gc_END).  
%param io_proc_hist_id For the insert operation, captures the new sequence-generated  
                       surrogate key ID and returns it to the caller to be used  
                       in the update/finish call when the process is done.  
%param i_tag_txt Text to use in "tagging" the row being inserted to fin_proc_hist  
  
------------------------------------------------------------------------------*/  
PROCEDURE tag_proc_hist  
(  
   i_sweep_dt      IN DATE,  
   i_op            IN VARCHAR2,  
   io_proc_hist_id IN OUT fin_proc_hist.proc_hist_id%TYPE,  
   i_status_cd     IN fin_proc_hist.status_cd%TYPE DEFAULT NULL,  
   i_tag_txt       IN fin_proc_hist.proc_txt%TYPE DEFAULT NULL  
) IS  
   PRAGMA AUTONOMOUS_TRANSACTION;  
BEGIN  
  
   logs.dbg(i_tag_txt);  
  
   IF (i_op = gc_start) THEN  
        
      INSERT INTO fin_proc_hist  
         (proc_hist_id  
         ,proc_nm  
         ,proc_dt  
         ,run_dt  
         ,end_dt  
         ,status_cd  
         ,proc_txt  
         ,client_id)  
      VALUES  
         (fin_proc_hist_seq.NEXTVAL  
         ,sweep_process_nm  
         ,i_sweep_dt  
         ,SYSDATE  
         ,NULL  
         ,NVL(i_status_cd,'RUNNING')  
         ,i_tag_txt  
         ,env.get_client_id)  
      RETURNING proc_hist_id INTO io_proc_hist_id;  
        
   ELSIF (i_op = gc_update) THEN  
        
      UPDATE fin_proc_hist  
         SET status_cd = NVL(i_status_cd,status_cd)  
            ,proc_txt = NVL(i_tag_txt,'Updated')  
       WHERE proc_hist_id = io_proc_hist_id;  
  
   ELSIF (i_op = gc_end) THEN  
        
      UPDATE fin_proc_hist  
         SET end_dt    = SYSDATE  
            ,status_cd = NVL(i_status_cd, 'SUCCESS')  
            ,proc_txt  = CASE WHEN i_tag_txt IS NULL THEN 'Ended'  
                              WHEN i_status_cd IN ('FAILED','FAILURE','ERROR') THEN  
                                    -- preserve the last action with the failure text  
                                    'Last action: ' || proc_txt || CHR(10)||  
                                    'Error msg: '||i_tag_txt  
                              ELSE i_tag_txt  
                         END  
            ,client_id = env.get_client_id  
       WHERE proc_hist_id = io_proc_hist_id;  
     
   END IF;  
   COMMIT;  
END tag_proc_hist;  
  
/**-----------------------------------------------------------------------------  
clean_prior_payments  
 Remove any rows for the given sweep date. We do not delete manual adjustment  
 rows on purpose. Users typically do not make adjustments until after the current  
 month's sweep has been finalized. Once finalized, this routine will never be  
 called for that month. The users can also, in theory, enter debits or credits  
 for future dates. We also do not want to delete those adjustments. So we only  
 delete adjustments made by the backend PL/SQL.  
------------------------------------------------------------------------------*/  
PROCEDURE clean_prior_payments(i_sweep_dt IN DATE)  
IS  
BEGIN  
   logs.dbg('Clean out payment history');  
   DELETE FROM fin_pay_hist  
    WHERE pay_dt = i_sweep_dt  
      AND mod_by = 'SYSTEM';  
END clean_prior_payments;  
  
/**-----------------------------------------------------------------------------  
clean_prior_sweep  
 Remove any data for the given sweep date.  
------------------------------------------------------------------------------*/  
PROCEDURE clean_prior_sweep(i_sweep_dt IN DATE)  
IS  
BEGIN  
   clean_prior_payments(i_sweep_dt);  
  
   logs.dbg('Clean out sweep tally');  
   DELETE FROM fin_sweep_tally  
    WHERE sweep_dt = i_sweep_dt;  
  
   logs.dbg('Clean out sweep missionary');  
   DELETE FROM fin_sweep_msny  
    WHERE sweep_dt = i_sweep_dt;  
      
  
END clean_prior_sweep;  
  
/**-----------------------------------------------------------------------------  
adj_msny_fund_type_id  
 Finds new changes to fund_type_id (can be overridden for unit in fin_unit_override  
 or changed for country on fin_poldivs) and applies them to the missionaries they  
 affect.  
------------------------------------------------------------------------------*/  
PROCEDURE adj_msny_fund_type_id  
IS  
   TYPE tr_fund_type_chg IS RECORD (  
      enamem_id INTEGER  
    , old_fund_type_id INTEGER  
    , new_fund_type_id INTEGER  
   );  
   TYPE tnt_changes IS TABLE OF tr_fund_type_chg INDEX BY PLS_INTEGER;  
   lnt_changes tnt_changes;  
BEGIN  
   SELECT enamem_id,  
          old_fund_type_id,  
          new_fund_type_id  
     BULK COLLECT INTO lnt_changes  
     FROM (  
      SELECT /*+ DYNAMIC_SAMPLING(fmsg 2) */  
              fmf.enamem_id  
              -- KEEP THESE COMMENTED-OUT COLUMNS AROUND FOR DEBUGGING  
              --,fsmg.legacy_miss_id  
              --,em.misstyp_id  
              --,em.procstat_id  
              --,DECODE(uo.unit_override_id,NULL,'N','Y') override  
              --,fmf.fund_unit_id  
              --,com.orglongname AS fund_unit_nm  
              --,ptc.primary_longname AS fund_unit_loc  
              --,ptc.country_longname AS fund_unit_cntry  
              ,fmf.fund_type_id AS old_fund_type_id  
               -- if unit not overriden, and not a US territory, use the country's funding method,  
               -- otherwise use the funding method of the US territory or override record  
              ,NVL(uo.fund_type_id, NVL(ust.fund_type_id,fp.fund_type_id)) AS new_fund_type_id  
          FROM fin_sweep_msny_gt fsmg -- limit changes to sweep set  
          JOIN fin_msny_funding fmf ON fmf.enamem_id = fsmg.enamem_id  
          JOIN enabled_members em ON em.enamem_id = fmf.enamem_id  
          JOIN fin_organization_mv com ON com.orgnbr = fmf.fund_unit_id  
          LEFT JOIN poldiv_to_country_mv ptc ON ptc.primary_poldivid = com.primarylocpoldivid  
          -- join to FUO to get unit's funding method overriding country setting  
          LEFT JOIN fin_unit_override uo ON uo.orgnbr = CASE  
                                                           WHEN fmf.allow_override_yn = 'Y' THEN   
                                                              com.orgnbr  
                                                           ELSE  
                                                              NULL -- prevent join if missionary is "locked" from unit overriding  
                                                        END  
          -- join to FP on country to get country's funding method  
          LEFT JOIN fin_poldivs fp ON fp.poldivid = ptc.country_poldivid  
          -- join inline view in case it is a US territory that must remain GMF  
          LEFT JOIN (  
             SELECT cpm.poldivid  
                    --,cpm.longname  
                   ,NVL(fp.fund_type_id, 2) AS fund_type_id  
               FROM col_poldiv_mv cpm  
               LEFT JOIN fin_poldivs fp ON fp.poldivid = cpm.poldivid  
              WHERE cpm.parpoldivid = 799  
                AND poltypedesctxt IN ('Territory', 'Country')  
          ) ust ON ust.poldivid = ptc.primary_poldivid  
         WHERE fmf.fund_unit_id IS NOT NULL -- all juniors should have funding unit  
           AND em.misstyp_id NOT IN (3,4,5)  
       )  
    WHERE new_fund_type_id IS NOT NULL -- don't take null as a valid new fund type id   
      AND old_fund_type_id <> new_fund_type_id;  
     
   IF (lnt_changes IS NOT NULL AND lnt_changes.COUNT > 0) THEN  
      FOR i IN lnt_changes.FIRST..lnt_changes.LAST LOOP  
         UPDATE fin_msny_funding  
            SET fund_type_id = lnt_changes(i).new_fund_type_id  
          WHERE enamem_id = lnt_changes(i).enamem_id;  
      END LOOP;  
   END IF;  
     
   logs.dbg('Updated fund_type_id for '||SQL%ROWCOUNT||' enabled members.');  
         
END adj_msny_fund_type_id;  
  
/**-----------------------------------------------------------------------------  
ins_sweep_msny  
 Captures missionaries active this month and their funding metadata.  
------------------------------------------------------------------------------*/  
PROCEDURE ins_sweep_msny(i_sweep_dt IN DATE)  
IS  
   l_created_dt DATE := SYSDATE;  
BEGIN  
   INSERT INTO fin_sweep_msny  
   (  
    sweep_dt  
   ,created_dt  
   ,msny_id  
   ,legacy_miss_id  
   ,enamem_id  
   ,procstat_id  
   ,msny_nm  
   ,misstyp_id  
   ,term_months  
   ,live_at_home_yn  
   ,mtc_dt  
   ,arriv_dt  
   ,release_dt  
   ,anniv_dt  
   ,mtc_id  
   ,mtc_cd  
   ,mtc_nm  
   ,call_type_id  
   ,comp_id  
   ,comp_nm  
   ,comp_asgloc_orgnbr  
   ,comp_asgloc_id  
   ,comp_asgloc_nm  
   ,comp_asgloc_orgtype_cd  
   ,comp_asgloc_orgtype_nm  
   ,comp_asgloc_country_id  
   ,comp_asgloc_country_nm  
   ,comp_asgloc_sens_flg          
   ,asgloc_orgnbr  
   ,asgloc_id  
   ,asgloc_nm  
   ,asgloc_orgtype_cd  
   ,asgloc_orgtype_nm  
   ,asgloc_area_orgnbr  
   ,asgloc_area_nm  
   ,asgloc_country_id  
   ,asgloc_country_nm  
   ,asgloc_sens_flg              
   ,within_msn_orgnbr  
   ,within_msn_asgloc_id  
   ,within_msn_nm  
   ,within_msn_sens_flg          
   ,eccl_org_orgnbr  
   ,eccl_org_nm  
   ,eccl_org_sens_flg            
   ,temp_comp_id  
   ,temp_comp_nm  
   ,temp_comp_asgloc_orgnbr  
   ,temp_comp_asgloc_id  
   ,temp_comp_asgloc_nm  
   ,temp_comp_asgloc_sens_flg    
   ,temp_asgloc_orgnbr  
   ,temp_asgloc_id  
   ,temp_asgloc_nm  
   ,temp_asgloc_sens_flg         
   ,temp_within_msn_orgnbr  
   ,temp_within_msn_asgloc_id  
   ,temp_within_msn_nm  
   ,temp_within_msn_sens_flg     
   ,temp_eccl_org_orgnbr  
   ,temp_eccl_org_nm  
   ,temp_eccl_org_sens_flg       
   ,fund_type_id  
   ,last_pmt_dt  
   ,blbl_month_yn  
   ,crnt_pmt_num  
   ,total_pmts  
   ,fund_unit_id  
   ,fund_unit_nm  
   ,fund_unit_sens_flg           
   ,fund_unit_stake_id  
   ,fund_unit_stake_nm  
   ,fund_unit_stake_sens_flg     
   ,fund_unit_area_orgnbr  
   ,fund_unit_area_nm  
   ,fund_unit_country_id  
   ,fund_unit_country_nm  
   ,fund_unit_currency_used  
   ,fund_unit_currency_cd  
   ,fund_unit_currency_nm  
   ,fund_unit_exchange_rt  
   ,fund_unit_equal_amt  
   ,fund_unit_commit_amt  
   ,fund_unit_local_amt  
   ,fund_unit_us_amt  
   ,msn_currency_used  
   ,msn_currency_cd  
   ,msn_currency_nm  
   ,msn_local_amt  
   ,msn_us_amt  
   ,msn_exchange_rt  
   ,msn_facs_num  
   )  
   SELECT i_sweep_dt AS sweep_dt  
         ,l_created_dt AS created_dt  
         ,gt.msny_id  
         ,gt.legacy_miss_id  
         ,gt.enamem_id  
         ,em.procstat_id  
         ,RTRIM(gt.last_nm || ', ' || gt.first_nm || ' ' || gt.middle_nm) AS msny_nm  
         ,em.misstyp_id  
         ,em.term_months  
         ,DECODE(alc.comp_housing_id,3,'Y','N') AS live_at_home_yn -- fixed during Sr. Exp work 
         ,em.enamem_mission_start_date AS mtc_dt  
         ,ca.asgmnt_start_date AS arriv_dt  
         ,em.enamem_release_date AS release_dt  
         ,em.enamem_anniversary_date AS anniv_dt  
         ,ca.call_mtcs_id AS mtc_id  
         ,ctom.orgshortname AS mtc_cd  
         ,tf.name AS mtc_nm  
         ,ca.call_type AS call_type_id  
         ,alc.comp_id AS comp_id  
         ,alc.comp_description AS comp_nm  
         ,cfom.orgnbr AS comp_asgloc_orgnbr  
         ,alc.asgloc_id AS comp_asgloc_id  
         ,cal.name AS comp_asgloc_nm  
         ,cfom.orgtypecode AS comp_asgloc_orgtype_cd  
         ,cfom.orgtypeshortdesctxt AS comp_asgloc_orgtype_nm  
         ,cpc.country_poldivid AS comp_asgloc_country_id  
         ,cpc.country_longname AS comp_asgloc_country_nm  
         ,cfom.orgsensitivitycode AS comp_asgloc_sens_flg  
         ,mfom.orgnbr AS asgloc_orgnbr  
         ,ca.call_aloc_id AS asgloc_id  
         ,mama.asgloc_name AS asgloc_nm  
         ,mfom.orgtypecode AS asgloc_orgtype_cd  
         ,mfom.orgtypeshortdesctxt AS asgloc_orgtype_nm  
         ,mama.area_orgnbr AS asgloc_area_orgnbr  
         ,mama.area_name AS asgloc_area_nm  
         ,mpc.country_poldivid AS asgloc_country_id  
         ,mpc.country_longname AS asgloc_country_nm  
         ,mfom.orgsensitivitycode AS asgloc_sens_flg  
         ,mama.wm_orgnbr AS within_msn_orgnbr  
         ,wmfom.asgloc_id AS within_msn_asgloc_id  
         ,mama.wm_orglongname AS within_msn_nm  
         ,wmfom.orgsensitivitycode AS within_msn_sens_flg  
         ,mama.eccl_orgnbr AS eccl_org_orgnbr  
         ,mama.eccl_orglongname AS eccl_org_nm  
         ,efom.orgsensitivitycode AS eccl_org_sens_flg  
         ,cta.call_comp_id as temp_comp_id  
         ,talc.comp_description AS temp_comp_nm  
         ,tcfom.orgnbr AS temp_comp_asgloc_orgnbr  
         ,talc.asgloc_id AS temp_comp_asgloc_id  
         ,tcal.name AS temp_comp_asgloc_nm  
         ,NVL(tcfom.orgsensitivitycode,0) AS temp_comp_asgloc_sens_flg  
         ,tmfom.orgnbr AS temp_asgloc_orgnbr  
         ,cta.call_aloc_id AS temp_asgloc_id  
         ,tama.asgloc_name AS temp_asgloc_nm  
         ,NVL(tmfom.orgsensitivitycode,0) AS temp_asgloc_sens_flg  
         ,tama.wm_orgnbr AS temp_within_msn_orgnbr  
         ,twmfom.asgloc_id AS temp_within_msn_asgloc_id  
         ,tama.wm_orglongname AS temp_within_msn_nm  
         ,NVL(twmfom.orgsensitivitycode,0) AS temp_within_msn_sens_flg  
         ,tama.eccl_orgnbr AS temp_eccl_org_orgnbr  
         ,tama.eccl_orglongname AS temp_eccl_org_nm  
         ,NVL(tefom.orgsensitivitycode,0) AS temp_eccl_org_sens_flg  
         ,fmf.fund_type_id  
         ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN lpm.last_pay_dt ELSE NULL END AS last_pmt_dt  
         ,CASE
            WHEN (lpm.last_pay_dt is NULL OR fmf.fund_type_id IS NULL)
               THEN 'N' -- not enough to go on
            WHEN (em.procstat_id = 11 AND TRUNC(em.enamem_mission_start_date, 'Month') > i_sweep_dt)
               THEN 'N' -- prefield that won't be in the MTC this month
            WHEN fmf.fund_type_id = 3 OR (MONTHS_BETWEEN(TRUNC(lpm.last_pay_dt,'Month'),i_sweep_dt) < 0)
               THEN 'N' -- unsupported or those whose last payment month has passed
            ELSE 'Y'  
          END AS blbl_month_yn  
         ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN NVL(gt.pmts_sent,0)+1 ELSE NULL END AS crnt_pmt_num  
         ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN NVL(gt.pmts_to_send,0) ELSE NULL END AS total_pmts  
         ,fmf.fund_unit_id AS fund_unit_id  
         ,ffom.orglongname AS fund_unit_nm  
         ,NVL(ffom.orgsensitivitycode,0) AS fund_unit_sens_flg  
         ,ffom.parentorgnbr AS fund_unit_stake_orgnbr  
         ,fstake.orglongname AS fund_unit_stake_nm  
         ,NVL(fstake.orgsensitivitycode,0) AS fund_unit_stake_sens_flg  
         ,ffom.areaorgnbr AS fund_unit_area_orgnbr  
         ,farea.orglongname AS fund_unit_area_nm  
         ,ffpc.country_poldivid AS fund_unit_country_id  
         ,ffpc.country_longname AS fund_unit_country_nm  
         ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN ffp.cfis_local_or_us_flag ELSE NULL END AS fund_unit_currency_used  
         ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN fcurr.iso_currency_code ELSE NULL END AS fund_unit_currency_cd  
         ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN fcurr.currency_name ELSE NULL END AS fund_unit_currency_nm  
         ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN fpcr.exch_rate ELSE NULL END AS fund_unit_exchange_rt  
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN ffp.finpol_equalized_pay_amt 
               WHEN fmf.fund_type_id IN (4,5) THEN alc.housing_exp
               ELSE NULL
          END AS fund_unit_equal_amt  
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN
                  -- add up commitment sources as reported by junior missionary  
                  NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0)
               WHEN fmf.fund_type_id IN (4,5) THEN
                  -- get housing commitment amount
                  fmf.housing_cmt_amt
          END AS fund_unit_commit_amt  
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN
                  -- Local Amount Due from Funding Unit = commitments if not equalized, otherwise the equalized amount
                  -- This formula for local and US amounts came from Legacy Pro*C file sweeptally.pc  
                  CASE  
                   -- check unit override table for equalization flag, before using eq flag for country  
                   WHEN NVL(fuo.equalized_yn, ffp.finpol_equalized_yn) = 'Y' THEN  
                      ffp.finpol_equalized_pay_amt  
                   ELSE   
                      LEAST(NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0),  
                            CASE  
                               WHEN (ffp.finpol_equalized_pay_amt IS NULL OR ffp.finpol_equalized_pay_amt = 0) THEN   
                                  NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0)  
                               ELSE  
                                  ffp.finpol_equalized_pay_amt  
                            END  
                      )  
                  END  
               WHEN fmf.fund_type_id = 4 THEN
                  CASE WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y' AND fmf.housing_exp_amt > 0) THEN
                          fmf.housing_exp_amt -- missionary housing expense is stored in local currency of funding unit
                       ELSE 0
                  END
               WHEN fmf.fund_type_id = 5 THEN
                  CASE
                     WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y') THEN
                        -- Compare housing commitment to housing expense. Take the lesser.
                        CASE WHEN fmf.housing_cmt_amt <= fmf.housing_exp_amt THEN
                                fmf.housing_cmt_amt
                             WHEN fmf.housing_cmt_amt > fmf.housing_exp_amt THEN
                                fmf.housing_exp_amt
                        END
                     ELSE 0
                  END
               ELSE 0
          END AS fund_unit_local_amt  
          -- USD amt due same as local, but multiplied by USD exchange rate  
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN
                  -- USD Amount Due from Funding Unit = commitments if not equalized, otherwise the equalized amount
                  -- This formula for local and US amounts came from Legacy Pro*C file sweeptally.pc  
                  ROUND(  
                     CASE  
                      -- check unit override table for equalization flag, before using eq flag for country  
                      WHEN NVL(fuo.equalized_yn, ffp.finpol_equalized_yn) = 'Y' THEN  
                         ffp.finpol_equalized_pay_amt  
                      ELSE   
                         LEAST(NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0),  
                               CASE  
                                  WHEN (ffp.finpol_equalized_pay_amt IS NULL OR ffp.finpol_equalized_pay_amt = 0) THEN   
                                     NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0)  
                                  ELSE  
                                     ffp.finpol_equalized_pay_amt  
                               END  
                         )  
                     END * fpcr.exch_rate 
                  )
               WHEN fmf.fund_type_id = 4 THEN
                  CASE WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y' AND fmf.housing_exp_amt > 0) THEN
                          ROUND(fmf.housing_exp_amt * fpcr.exch_rate)-- missionary housing expense in terms of USD
                       ELSE 0
                  END
               WHEN fmf.fund_type_id = 5 THEN
                  CASE
                     WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y') THEN
                        -- Compare commitment to housing expense and take the lesser
                        ROUND(
                           CASE WHEN fmf.housing_cmt_amt <= fmf.housing_exp_amt THEN
                                   fmf.housing_cmt_amt
                                WHEN fmf.housing_cmt_amt > fmf.housing_exp_amt THEN
                                   fmf.housing_exp_amt
                           END * fpcr.exch_rate
                        )
                     ELSE 0
                  END
               ELSE 0
          END AS fund_unit_us_amt  
         ,mfp.cfis_local_or_us_flag AS msn_currency_used  
         ,mcurr.iso_currency_code AS msn_currency_cd  
         ,mcurr.currency_name AS msn_currency_nm  
          -- this formula for local and US amounts came from Legacy Pro*C file sweeptally.pc  
         ,CASE WHEN mfp.cfis_local_or_us_flag = 'L' THEN mfa.base_amt ELSE ROUND(mfa.base_amt / mpcr.exch_rate) END AS msn_local_amt  
         ,CASE WHEN mfp.cfis_local_or_us_flag = 'L' THEN ROUND(mfa.base_amt * mpcr.exch_rate) ELSE mfa.base_amt END AS msn_us_amt  
         ,mpcr.exch_rate AS msn_exchange_rt  
         ,mfa.gl_unit_num AS msn_facs_num  
     FROM fin_sweep_msny_gt gt -- this is the filter for what gets swept  
          JOIN enabled_members em ON em.enamem_id = gt.enamem_id  
          JOIN current_perm_assignment_vw ca ON ca.call_emem_id = gt.enamem_id  
          JOIN fin_msny_funding fmf ON fmf.enamem_id = gt.enamem_id  
          JOIN (  
                 SELECT g.msny_id,   
                        CASE  
                           -- How many days beyond planned release is the actual release?  
                           WHEN (e.enamem_release_date IS NULL OR e.enamem_anniversary_date IS NULL) THEN  
                              NULL -- return empty value to be picked up by error module  
                           WHEN (e.enamem_release_date - e.enamem_anniversary_date) > 14 THEN
                              ADD_MONTHS(e.enamem_anniversary_date, 
                                         (TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))
                                        +(CASE
                                             WHEN e.enamem_release_date - (ADD_MONTHS(e.enamem_anniversary_date,TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))) > 14
                                                THEN 0
                                             ELSE -1
                                          END)
                                        )  
                           WHEN (e.enamem_release_date - e.enamem_anniversary_date) < -14 THEN  
                              ADD_MONTHS(e.enamem_anniversary_date, 
                                         (TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))
                                        -(CASE
                                             WHEN e.enamem_release_date - (ADD_MONTHS(e.enamem_anniversary_date,TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))) < -14
                                                THEN 2
                                             ELSE 1
                                          END)
                                        )  
                           ELSE  
                              ADD_MONTHS(e.enamem_anniversary_date,-1)  
                        END AS last_pay_dt  
                 FROM fin_sweep_msny_gt g  
                 JOIN enabled_members e ON e.enamem_id = g.enamem_id  
               ) lpm  
               ON lpm.msny_id = gt.msny_id  
          JOIN asg_loc_components alc ON alc.comp_id = ca.call_comp_id  
          -- component-level asgloc info  
          JOIN fin_organization_mv cfom ON cfom.asgloc_id = alc.asgloc_id  
          JOIN assignment_locations cal ON cal.asgloc_id = alc.asgloc_id  
          JOIN poldiv_to_country_mv cpc ON cpc.primary_poldivid = cfom.primarylocpoldivid  
          -- "parent" asgloc info  
          JOIN fin_organization_mv mfom ON mfom.asgloc_id = ca.call_aloc_id  
          JOIN asgloc_msn_area_mv mama ON mama.asgloc_orgnbr = mfom.orgnbr  
          JOIN poldiv_to_country_mv mpc ON mpc.primary_poldivid = mfom.primarylocpoldivid  
          -- "within mission" info  
          JOIN fin_organization_mv wmfom ON wmfom.orgnbr = mama.wm_orgnbr  
          -- ecclesiastical info  
          JOIN fin_organization_mv efom ON efom.orgnbr = mama.eccl_orgnbr  
          -- joins to get financial info for the mission   
          JOIN fin_poldivs mfp ON mfp.poldivid = mpc.country_poldivid  
          JOIN mdm_currency mcurr ON mcurr.iso_currency_code = mfp.finpol_iso_currency_code  
          JOIN fin_exch_rate mpcr ON mpcr.from_cur = mcurr.iso_currency_code
          JOIN fin_asglocs mfa ON mfa.asgloc_id = mfom.asgloc_id  
          -- start of outer joins ---------------------------------------------
          LEFT JOIN training_facilities tf ON tf.trnfac_id = ca.call_mtcs_id -- many MTC IDs are null, some 0  
          LEFT JOIN col_training_orgs_mv ctom ON ctom.orgnbr = tf.col_orgnbr -- must left join as well if MTC was null  
          LEFT JOIN fin_organization_mv ffom ON ffom.orgnbr = fmf.fund_unit_id -- some test missionaries won't have entries in fin_organization_mv  
          LEFT JOIN fin_organization_mv fstake ON fstake.orgnbr = ffom.parentorgnbr -- need long name of stake  
          LEFT JOIN fin_organization_mv farea ON farea.orgnbr = ffom.areaorgnbr -- need long name of area  
          LEFT JOIN poldiv_to_country_mv fpc ON fpc.primary_poldivid = ffom.primarylocpoldivid  
          LEFT JOIN fin_unit_override fuo  
               ON fuo.orgnbr = CASE  
                                  WHEN fmf.allow_override_yn = 'Y' THEN ffom.orgnbr  
                                  ELSE NULL -- remove ability to join to override table  
                               END  
          LEFT JOIN fin_poldivs ffp -- join for funding unit country financial attributes  
               ON ffp.poldivid = CASE   
                                    WHEN fuo.orgnbr IS NULL THEN fpc.country_poldivid   
                                    ELSE fuo.poldivid_of_equalized_amt   
                                 END  
          -- Now that fin_poldivs is joined, perhaps on redirected country from poldivid_of_equalized_amt,  
          -- re-join to poldiv_to_country_mv to get country info.  
          LEFT JOIN poldiv_to_country_mv ffpc ON ffpc.primary_poldivid = ffp.poldivid  
          LEFT JOIN mdm_currency fcurr ON fcurr.iso_currency_code = ffp.finpol_iso_currency_code
          LEFT JOIN fin_exch_rate fpcr ON fpcr.from_cur = fcurr.iso_currency_code  
          -- Plethora of outer joins to get possible temp assignment info   
          LEFT JOIN (  
             -- current_temp_assignment_vw didn't work since it internally used SYSDATE to filter. We  
             -- need the filter to use the sweep date, the 1st of the month.  
             SELECT call_id, call_aloc_id, call_comp_id, call_emem_id, asgmnt_start_date, asgmnt_end_date  
              FROM napi_assignment_mstr
             WHERE call_type in (2,3) 
               AND asgmnt_start_date <= i_sweep_dt  
               AND (asgmnt_end_date > i_sweep_dt OR asgmnt_end_date IS NULL)  
               AND status <> 7  
          ) cta ON cta.call_emem_id = em.enamem_id  
          LEFT JOIN asg_loc_components talc ON talc.comp_id = cta.call_comp_id  
          LEFT JOIN fin_organization_mv tcfom ON tcfom.asgloc_id = talc.asgloc_id  
          LEFT JOIN assignment_locations tcal ON tcal.asgloc_id = talc.asgloc_id  
          LEFT JOIN fin_organization_mv tmfom ON tmfom.asgloc_id = cta.call_aloc_id  
          LEFT JOIN asgloc_msn_area_mv tama ON tama.asgloc_id = cta.call_aloc_id  
          LEFT JOIN fin_organization_mv twmfom ON twmfom.orgnbr = tama.wm_orgnbr  
          LEFT JOIN fin_organization_mv tefom ON tefom.orgnbr = tama.eccl_orgnbr  
          ;  
  
END ins_sweep_msny;  
  
/**-----------------------------------------------------------------------------  
ins_sweep_tally  
 Sums counts of missionary types and status by month, mission and component.  
------------------------------------------------------------------------------*/  
PROCEDURE ins_sweep_tally(i_sweep_dt IN DATE)  
IS  
   l_created_dt DATE := SYSDATE;  
BEGIN  
     
   INSERT INTO fin_sweep_tally  
   (   
     sweep_dt, created_dt  
   , comp_id, comp_nm  
   , comp_asgloc_orgnbr, comp_asgloc_orgtype_cd, comp_asgloc_orgtype_nm  
   , comp_asgloc_id, comp_asgloc_nm, comp_asgloc_sens_flg  
   , asgloc_orgnbr, asgloc_id, asgloc_nm  
   , asgloc_area_orgnbr, asgloc_area_nm  
   , asgloc_country_id, asgloc_country_nm, asgloc_sens_flg  
   , within_msn_orgnbr, within_msn_asgloc_id, within_msn_nm, within_msn_sens_flg  
   , eccl_org_orgnbr, eccl_org_nm, eccl_org_sens_flg  
   , msn_currency_used, msn_currency_cd, msn_currency_nm, msn_exchange_rt  
   , msn_local_amt, msn_us_amt, msn_facs_num  
   , pos_spcl_id, prim_pos_nm, prim_pos_pros_only_yn  
   , rspbl_org_nm, rspbl_orgnbr  
   , lang_id, lang_nm  
   , complement_elders, complement_sisters, complement_couples, complement_selders, complement_ssisters, prefield_elders  
   , prefield_sisters, prefield_couples, prefield_selders, prefield_ssisters  
   , mtc_elders, mtc_sisters, mtc_couples, mtc_selders, mtc_ssisters  
   , infield_elders, infield_sisters, infield_couples, infield_selders, infield_ssisters  
   , temp_elders, temp_sisters, temp_couples, temp_selders, temp_ssisters  
   , away_elders, away_sisters, away_couples, away_selders, away_ssisters  
   , lah_couples, lah_selders, lah_ssisters  
   )  
   SELECT * FROM (  
    WITH finsm AS  
       (SELECT * FROM fin_sweep_msny WHERE sweep_dt = i_sweep_dt)   
    SELECT   
      sweep_dt, l_created_dt  
    , comp_id, comp_nm  
    , comp_asgloc_orgnbr, comp_asgloc_orgtype_cd, comp_asgloc_orgtype_nm  
    , comp_asgloc_id, comp_asgloc_nm, comp_asgloc_sens_flg  
    , asgloc_orgnbr, asgloc_id, asgloc_nm  
    , asgloc_area_orgnbr, asgloc_area_nm  
    , asgloc_country_id, asgloc_country_nm, asgloc_sens_flg  
    , within_msn_orgnbr, within_msn_asgloc_id, within_msn_nm, within_msn_sens_flg  
    , eccl_org_orgnbr, eccl_org_nm, eccl_org_sens_flg  
    , msn_currency_used, msn_currency_cd, msn_currency_nm, msn_exchange_rt  
    , msn_local_amt, msn_us_amt, msn_facs_num  
    , pos_spcl_id, prim_pos_nm, prim_pos_pros_only_yn  
    , rspbl_org_nm, rspbl_orgnbr  
    , lang_id, lang_nm  
    , complement_elders, complement_sisters, complement_couples, complement_selders, complement_ssisters  
    , prefield_elders, prefield_sisters, prefield_couples, prefield_selders, prefield_ssisters  
    , mtc_elders, mtc_sisters, mtc_couples, mtc_selders, mtc_ssisters  
    , infield_elders, infield_sisters, infield_couples, infield_selders, infield_ssisters  
    , CASE WHEN misstyp_id = 1 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_elders  
    , CASE WHEN misstyp_id = 2 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_sisters  
    , CASE WHEN misstyp_id = 3 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_couples  
    , CASE WHEN misstyp_id = 4 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_selders  
    , CASE WHEN misstyp_id = 5 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_ssisters  
    , away_elders, away_sisters, away_couples, away_selders, away_ssisters  
    , lah_couples, lah_selders, lah_ssisters  
    FROM (  
       SELECT   
         fsm.sweep_dt  
       , fsm.comp_id, fsm.comp_nm  
       , fsm.comp_asgloc_orgnbr, fsm.comp_asgloc_orgtype_cd, fsm.comp_asgloc_orgtype_nm  
       , fsm.comp_asgloc_id, fsm.comp_asgloc_nm, fsm.comp_asgloc_sens_flg  
       , fsm.asgloc_orgnbr, fsm.asgloc_id, fsm.asgloc_nm  
       , fsm.asgloc_area_orgnbr, fsm.asgloc_area_nm  
       , fsm.asgloc_country_id, fsm.asgloc_country_nm, fsm.asgloc_sens_flg  
       , fsm.within_msn_orgnbr, fsm.within_msn_asgloc_id, fsm.within_msn_nm, fsm.within_msn_sens_flg  
       , fsm.eccl_org_orgnbr, fsm.eccl_org_nm, fsm.eccl_org_sens_flg  
       , fsm.msn_currency_used, fsm.msn_currency_cd, fsm.msn_currency_nm, fsm.msn_exchange_rt  
       , fsm.msn_local_amt, fsm.msn_us_amt, fsm.msn_facs_num  
       , acpv.posspc_id pos_spcl_id, acpv.primary_pos_name prim_pos_nm, acpv.primary_pos_pros_only prim_pos_pros_only_yn  
       , acpv.responsible_org rspbl_org_nm, acpv.responsible_orgnbr rspbl_orgnbr  
       , acpv.lang_id, acpv.language_name lang_nm  
       , alc.misstyp_id  
       , CASE WHEN alc.misstyp_id = 1 THEN alc.comp_num ELSE 0 END AS complement_elders  
       , CASE WHEN alc.misstyp_id = 2 THEN alc.comp_num ELSE 0 END AS complement_sisters  
       , CASE WHEN alc.misstyp_id = 3 THEN alc.comp_num ELSE 0 END AS complement_couples  
       , CASE WHEN alc.misstyp_id = 4 THEN alc.comp_num ELSE 0 END AS complement_selders  
       , CASE WHEN alc.misstyp_id = 5 THEN alc.comp_num ELSE 0 END AS complement_ssisters  
       , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_elders  
       , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_sisters  
       , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_couples  
       , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_selders  
       , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_ssisters  
       , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_elders  
       , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_sisters  
       , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_couples  
       , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_selders  
       , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_ssisters  
       , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_elders  
       , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_sisters  
       , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_couples  
       , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_selders  
       , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_ssisters  
       , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_elders  
       , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_sisters  
       , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_couples  
       , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_selders  
       , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_ssisters  
       , SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND alc.misstyp_id = 3 THEN 1 ELSE 0 END) AS lah_couples  
       , SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND alc.misstyp_id = 4 THEN 1 ELSE 0 END) AS lah_selders  
       , SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND alc.misstyp_id = 5 THEN 1 ELSE 0 END) AS lah_ssisters  
       FROM finsm fsm  
            JOIN asg_loc_components alc ON alc.comp_id = fsm.comp_id  
            JOIN asgloc_comp_pos_vw acpv ON acpv.comp_id = fsm.comp_id -- join for position info  
       GROUP BY  
         fsm.sweep_dt  
       , fsm.comp_id, fsm.comp_nm  
       , fsm.comp_asgloc_orgnbr, fsm.comp_asgloc_orgtype_cd, fsm.comp_asgloc_orgtype_nm  
       , fsm.comp_asgloc_id, fsm.comp_asgloc_nm, fsm.comp_asgloc_sens_flg  
       , fsm.asgloc_orgnbr, fsm.asgloc_id, fsm.asgloc_nm  
       , fsm.asgloc_area_orgnbr, fsm.asgloc_area_nm  
       , fsm.asgloc_country_id, fsm.asgloc_country_nm, fsm.asgloc_sens_flg  
       , fsm.within_msn_orgnbr, fsm.within_msn_asgloc_id, fsm.within_msn_nm, fsm.within_msn_sens_flg  
       , fsm.eccl_org_orgnbr, fsm.eccl_org_nm, fsm.eccl_org_sens_flg  
       , fsm.msn_currency_used, fsm.msn_currency_cd, fsm.msn_currency_nm, fsm.msn_exchange_rt  
       , fsm.msn_local_amt, fsm.msn_us_amt, fsm.msn_facs_num  
       , acpv.posspc_id, acpv.primary_pos_name, acpv.primary_pos_pros_only  
       , acpv.responsible_org, acpv.responsible_orgnbr  
       , acpv.lang_id, acpv.language_name  
       , alc.misstyp_id, alc.comp_num  
   ) fsm  
   LEFT OUTER JOIN (SELECT temp_comp_id, COUNT(*) msny_count FROM finsm GROUP BY temp_comp_id) tfsw ON tfsw.temp_comp_id = fsm.comp_id)  
   ;  
  
  
   -- Now grab those missionaries who are temporarily assigned to components  
   -- not used by any permanent assignment in the previous query from fin_sweep_msny.  
   INSERT INTO fin_sweep_tally  
   (   
     sweep_dt, created_dt  
   , comp_id, comp_nm  
   , comp_asgloc_orgnbr, comp_asgloc_orgtype_cd, comp_asgloc_orgtype_nm  
   , comp_asgloc_id, comp_asgloc_nm, comp_asgloc_sens_flg  
   , asgloc_orgnbr, asgloc_id, asgloc_nm  
   , asgloc_area_orgnbr, asgloc_area_nm  
   , asgloc_country_id, asgloc_country_nm, asgloc_sens_flg  
   , within_msn_orgnbr, within_msn_asgloc_id, within_msn_nm, within_msn_sens_flg  
   , eccl_org_orgnbr, eccl_org_nm, eccl_org_sens_flg  
   , msn_currency_used, msn_currency_cd, msn_currency_nm, msn_exchange_rt  
   , msn_local_amt, msn_us_amt, msn_facs_num  
   , pos_spcl_id, prim_pos_nm, prim_pos_pros_only_yn  
   , rspbl_org_nm, rspbl_orgnbr  
   , lang_id, lang_nm  
   , complement_elders, complement_sisters, complement_couples, complement_selders, complement_ssisters, prefield_elders  
   , prefield_sisters, prefield_couples, prefield_selders, prefield_ssisters  
   , mtc_elders, mtc_sisters, mtc_couples, mtc_selders, mtc_ssisters  
   , infield_elders, infield_sisters, infield_couples, infield_selders, infield_ssisters  
   , temp_elders, temp_sisters, temp_couples, temp_selders, temp_ssisters  
   , away_elders, away_sisters, away_couples, away_selders, away_ssisters  
   , lah_couples, lah_selders, lah_ssisters  
   )  
   SELECT * FROM   
   (  
   WITH ot AS -- "ot" stands for orphaned temporaries  
      (  
         SELECT fsm.temp_comp_id, fsm.temp_comp_nm,  
                fsm.temp_comp_asgloc_orgnbr, fsm.temp_comp_asgloc_id, fsm.temp_comp_asgloc_nm, fsm.temp_comp_asgloc_sens_flg,  
                fsm.temp_asgloc_orgnbr, fsm.temp_asgloc_id, fsm.temp_asgloc_nm, fsm.temp_asgloc_sens_flg,  
                fsm.temp_within_msn_orgnbr, fsm.temp_within_msn_asgloc_id, fsm.temp_within_msn_nm, fsm.temp_within_msn_sens_flg,  
                fsm.temp_eccl_org_orgnbr, fsm.temp_eccl_org_nm, fsm.temp_eccl_org_sens_flg,  
                SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND fsm.misstyp_id = 3 THEN 1 ELSE 0 END) AS lah_couples,  
                SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND fsm.misstyp_id = 4 THEN 1 ELSE 0 END) AS lah_selders,  
                SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND fsm.misstyp_id = 5 THEN 1 ELSE 0 END) AS lah_ssisters,  
                COUNT(*) AS cnt  
           FROM fin_sweep_msny fsm  
          WHERE fsm.sweep_dt = i_sweep_dt  
            AND fsm.temp_comp_id IS NOT NULL  
            AND NOT EXISTS (SELECT NULL  
                   FROM fin_sweep_tally fst  
                  WHERE fst.sweep_dt = i_sweep_dt  
                    AND fst.comp_id = fsm.temp_comp_id)  
         GROUP BY   
                fsm.temp_comp_id, fsm.temp_comp_nm,  
                fsm.temp_comp_asgloc_orgnbr, fsm.temp_comp_asgloc_id, fsm.temp_comp_asgloc_nm, fsm.temp_comp_asgloc_sens_flg,  
                fsm.temp_asgloc_orgnbr, fsm.temp_asgloc_id, fsm.temp_asgloc_nm, fsm.temp_asgloc_sens_flg,  
                fsm.temp_within_msn_orgnbr, fsm.temp_within_msn_asgloc_id, fsm.temp_within_msn_nm, fsm.temp_within_msn_sens_flg,  
                fsm.temp_eccl_org_orgnbr, fsm.temp_eccl_org_nm, fsm.temp_eccl_org_sens_flg  
      )  
   SELECT  
     i_sweep_dt, l_created_dt  
   , ot.temp_comp_id AS comp_id, ot.temp_comp_nm AS comp_nm  
   , ot.temp_comp_asgloc_orgnbr AS comp_asgloc_orgnbr, cfom.orgtypecode AS comp_asgloc_orgtype_cd, cfom.orgtypeshortdesctxt AS comp_asgloc_orgtype_nm  
   , ot.temp_comp_asgloc_id AS comp_asgloc_id, ot.temp_comp_asgloc_nm  AS comp_asgloc_nm, ot.temp_comp_asgloc_sens_flg AS comp_asgloc_sens_flg  
   , ot.temp_asgloc_orgnbr AS asgloc_orgnbr, ot.temp_asgloc_id AS asgloc_id, ot.temp_asgloc_nm AS asgloc_nm  
   , acom.areaorgnbr AS asgloc_area_orgnbr, acom.areamediumname AS asgloc_area_nm  
   , apc.country_poldivid asgloc_country_id, apc.country_longname AS asgloc_country_nm, ot.temp_asgloc_sens_flg AS asgloc_sens_flg  
   , ot.temp_within_msn_orgnbr AS within_msn_orgnbr, ot.temp_within_msn_asgloc_id AS within_msn_asgloc_id, ot.temp_within_msn_nm AS within_msn_nm, ot.temp_within_msn_sens_flg AS within_msn_sens_flg  
   , ot.temp_eccl_org_orgnbr AS eccl_org_orgnbr, ot.temp_eccl_org_nm AS eccl_org_nm, ot.temp_eccl_org_sens_flg AS eccl_org_sens_flg  
   , afp.cfis_local_or_us_flag AS msn_currency_used, afp.finpol_iso_currency_code AS msn_currency_cd  
   , acurr.currency_desc AS msn_currency_nm, apcr.exch_rate AS msn_exchange_rt  
   , CASE WHEN afp.cfis_local_or_us_flag = 'L' THEN afa.base_amt ELSE ROUND(afa.base_amt / apcr.exch_rate) END AS msn_local_amt  
   , CASE WHEN afp.cfis_local_or_us_flag = 'L' THEN ROUND(afa.base_amt * apcr.exch_rate) ELSE afa.base_amt END AS msn_us_amt  
   , afa.gl_unit_num AS msn_facs_num  
   , acpv.posspc_id pos_spcl_id, acpv.primary_pos_name prim_pos_nm, acpv.primary_pos_pros_only prim_pos_pros_only_yn  
   , acpv.responsible_org rspbl_org_nm, acpv.responsible_orgnbr rspbl_orgnbr  
   , acpv.lang_id, acpv.language_name lang_nm  
   , CASE WHEN alc.misstyp_id = 1 THEN alc.comp_num ELSE 0 END AS complement_elders  
   , CASE WHEN alc.misstyp_id = 2 THEN alc.comp_num ELSE 0 END AS complement_sisters  
   , CASE WHEN alc.misstyp_id = 3 THEN alc.comp_num ELSE 0 END AS complement_couples  
   , CASE WHEN alc.misstyp_id = 4 THEN alc.comp_num ELSE 0 END AS complement_selders  
   , CASE WHEN alc.misstyp_id = 5 THEN alc.comp_num ELSE 0 END AS complement_ssisters  
   -- Missionaries with active temp assignments will always be infield. They can have planned  
   -- temp assignments while prefield or MTC, but those assignments won't show up  
   -- in current temp assignment inline view, which is used in the initial insert to fin_sweep_msny  
   , 0 AS prefield_elders  
   , 0 AS prefield_sisters  
   , 0 AS prefield_couples  
   , 0 AS prefield_selders  
   , 0 AS prefield_ssisters  
   , 0 AS mtc_elders  
   , 0 AS mtc_sisters  
   , 0 AS mtc_couples  
   , 0 AS mtc_selders  
   , 0 AS mtc_ssisters  
   , 0 AS infield_elders  
   , 0 AS infield_sisters  
   , 0 AS infield_couples  
   , 0 AS infield_selders  
   , 0 AS infield_ssisters  
   , CASE WHEN alc.misstyp_id = 1 THEN ot.cnt ELSE 0 END AS temp_elders  
   , CASE WHEN alc.misstyp_id = 2 THEN ot.cnt ELSE 0 END AS temp_sisters  
   , CASE WHEN alc.misstyp_id = 3 THEN ot.cnt ELSE 0 END AS temp_couples  
   , CASE WHEN alc.misstyp_id = 4 THEN ot.cnt ELSE 0 END AS temp_selders  
   , CASE WHEN alc.misstyp_id = 5 THEN ot.cnt ELSE 0 END AS temp_ssisters  
   , 0 AS away_elders  
   , 0 AS away_sisters  
   , 0 AS away_couples  
   , 0 AS away_selders  
   , 0 AS away_ssisters  
   , ot.lah_couples  
   , ot.lah_selders  
   , ot.lah_ssisters  
   FROM ot  
   JOIN assignment_locations al  
        ON al.asgloc_id = ot.temp_asgloc_id  
   JOIN asg_loc_components alc  
        ON alc.comp_id = ot.temp_comp_id  
   JOIN assignment_locations cal  
        ON cal.asgloc_id = alc.asgloc_id  
   JOIN fin_organization_mv cfom -- join for component-level assignment location info  
        ON cfom.orgnbr = cal.asgloc_col_orgnbr  
   JOIN asgloc_comp_pos_vw acpv -- join for position info  
        ON acpv.comp_id = ot.temp_comp_id  
   JOIN fin_organization_mv acom -- join for assignment location info  
        ON acom.orgnbr = al.asgloc_col_orgnbr  
   JOIN poldiv_to_country_mv apc -- join for assignment location country  
        ON apc.primary_poldivid = acom.primarylocpoldivid  
   JOIN fin_poldivs afp -- join for assignment location country financial attributes  
        ON afp.poldivid = apc.country_poldivid  
   JOIN currency_type_vw acurr  
        ON acurr.iso_currency_code = afp.finpol_iso_currency_code  
   JOIN fin_exch_rate apcr  
        ON apcr.from_cur = acurr.iso_currency_code  
   JOIN fin_asglocs afa  
        ON afa.asgloc_id = al.asgloc_id  
   );  
        
END ins_sweep_tally;  
  
/**-----------------------------------------------------------------------------  
ins_pay_hist  
 Captures records for each missionary funded by a funding unit.  
------------------------------------------------------------------------------*/  
PROCEDURE ins_pay_hist(i_sweep_dt IN DATE)  
IS  
BEGIN  
  
   INSERT INTO fin_pay_hist  
      (msny_id  
      ,legacy_miss_id  
      ,fund_unit_id  
      ,fund_type_id  
      ,pay_dt  
      ,amount  
      ,pay_nbr  
      ,dr_cr_code  
      ,cfar_comments  
      ,mod_by  
      ,mod_dt  
      ,adj_yn  
      ,currency_cd)  
      SELECT msny_id  
            ,legacy_miss_id  
            ,fund_unit_id  
            ,fund_type_id  
            ,sweep_dt  
            ,fund_unit_local_amt  
            ,crnt_pmt_num  
            ,'DR'  
            ,'Monthly Sweep'  
            ,'SYSTEM'  
            ,SYSDATE  
            ,NULL AS adj_yn -- system-created pay hist rows use NULL. Y reserved for human adjustments.  
            ,fund_unit_currency_cd AS currency_cd  
        FROM fin_sweep_msny  
       WHERE sweep_dt = i_sweep_dt  
         AND blbl_month_yn = 'Y'
         AND (misstyp_id IN (1,2) OR (misstyp_id = 3 AND fund_unit_local_amt > 0 ))
       ORDER BY fund_unit_id, call_type_id, procstat_id DESC, msny_id;  
                               
END ins_pay_hist;  
  
  
/**-----------------------------------------------------------------------------  
run_error_check  
 Creates error records in FIN_SWEEP_MSNY_ERR for each anomalous condition it   
 finds.  
   
 The error table is cleared out each time the error check is run. Then the   
 various checks are run, writing one record for each missionary sporting the   
 error condition.  
   
 Will be run each time the monthly sweep is run.  
 Can be run independently if needed.   
 Can even be run after a sweep has been finalized (assuming a fresh view of this  
 month's acceptable errors is required).  
   
%param i_sweep_dt  The effective date of the run. The provided date will  
                   be truncated to the first of the month.  
------------------------------------------------------------------------------*/  
PROCEDURE run_error_check(i_sweep_dt IN DATE DEFAULT SYSDATE)  
IS  
   l_sweep_dt DATE := TRUNC(NVL(i_sweep_dt,SYSDATE),'Month');  
   l_marker INTEGER := 0;  
   l_count INTEGER := 0;  
BEGIN  
   logs.dbg('Clean out the error table');  
   DELETE FROM fin_sweep_msny_err;  
     
   -- If being called independently, populate the sweep set first  
   SELECT COUNT(*) INTO l_count FROM fin_sweep_msny_gt;  
   IF (l_count = 0) THEN  
      load_sweep_set(l_sweep_dt);  
   END IF;  
  
   -- missing mission start date  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Missing mission start (MTC) date.'  
            ,'ENABLED_MEMBER_MSTR'  
            ,'ENAMEM_MISSION_START_DATE'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
             ON em.enamem_id = gt.enamem_id  
       WHERE em.enamem_mission_start_date IS NULL;  
     
   -- missing release date  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Missing release date.'  
            ,'ENABLED_MEMBER_MSTR'  
            ,'ENAMEM_RELEASE_DATE'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
             ON em.enamem_id = gt.enamem_id  
       WHERE em.enamem_release_date IS NULL;  
     
   -- missing anniversary date  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Missing anniversary date.'  
            ,'ENABLED_MEMBER_MSTR'  
            ,'ENAMEM_ANNIVERSARY_DATE'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
             ON em.enamem_id = gt.enamem_id  
       WHERE em.enamem_anniversary_date IS NULL;  
         
   -- anniversary date before mission start date  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Anniversary before start date.'  
            ,'ENABLED_MEMBER_MSTR'  
            ,'ENAMEM_ANNIVERSARY_DATE'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
             ON em.enamem_id = gt.enamem_id  
       WHERE em.enamem_anniversary_date < em.enamem_mission_start_date;  
         
   -- sweep missionaries that don't have assignments  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm)  
      SELECT gt.msny_id  
            ,'Missing assignment.'  
            ,'NAPI_ASSIGNMENT_MSTR'  
        FROM fin_sweep_msny_gt gt  
        LEFT JOIN current_perm_assignment_vw ca  
             ON ca.call_emem_id = gt.enamem_id  
       WHERE ca.call_emem_id IS NULL;  
         
   -- missing home unit              
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT m.ID AS msny_id  
            ,'Missing home unit.'  
            ,'ENABLED_MEMBER_MSTR'  
            ,'ENAMEM_HOME_UNIT_ID'  
        FROM missionaries m  
        JOIN enabled_members em  
             ON em.enamem_id = m.enamem_id  
       WHERE em.procstat_id IN (11, 12, 13, 14)  
         AND em.enamem_release_date >= l_sweep_dt  
         AND em.enamem_home_unit_id IS NULL;  
  
   -- has assignment, but start date is missing  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Missing assignment start date.'  
            ,'NAPI_ASSIGNMENT_MSTR'  
            ,'ASGMNT_START_DATE'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
             ON em.enamem_id = gt.enamem_id  
        JOIN current_perm_assignment_vw ca  
             ON ca.call_emem_id = em.enamem_id  
       WHERE ca.asgmnt_start_date IS NULL;  
  
   -- has assignment, but end date is missing  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Missing assignment end date.'  
            ,'NAPI_ASSIGNMENT_MSTR'  
            ,'ASGMNT_END_DATE'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
             ON em.enamem_id = gt.enamem_id  
        JOIN current_perm_assignment_vw ca  
             ON ca.call_emem_id = em.enamem_id  
       WHERE ca.asgmnt_end_date IS NULL;  
            
   -- missionaries that don't have funding record  
   -- all missionaries going on, or on a mission should have a funding record  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm)  
      SELECT gt.msny_id  
            ,'Missing funding record.'  
            ,'FIN_MSNY_FUNDING'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
             ON em.enamem_id = gt.enamem_id  
        LEFT JOIN fin_msny_funding fmf  
             ON fmf.enamem_id = em.enamem_id  
       WHERE fmf.enamem_id IS NULL;  
         
   -- has funding record, but don't have a funding unit  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Missing funding unit.'  
            ,'FIN_MSNY_FUNDING'  
            ,'FUND_UNIT_ID'  
        FROM fin_sweep_msny_gt gt  
        JOIN enabled_members em  
          ON em.enamem_id = gt.enamem_id  
        JOIN fin_msny_funding fmf  
          ON fmf.enamem_id = em.enamem_id  
       WHERE fmf.fund_type_id IN (1, 2, 4, 5)  
         AND fmf.fund_unit_id IS NULL;  
     
   -- Missionary is assigned to inactive (Deactivated) components  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT fsm.msny_id  
            ,'Missionary assigned to deactivated component ' || fsm.comp_nm ||  
             ' in ' || fsm.comp_asgloc_nm || ' (comp_id=' || fsm.comp_id || ').'  
            ,'NAPI_ASSIGNMENT_MSTR'  
            ,'COMP_ID'  
        FROM fin_sweep_msny fsm  
        JOIN asg_loc_components alc  
          ON alc.comp_id = fsm.comp_id  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND alc.compsta_id = 3;  
     
   -- Missionary is assigned to inactive (Discontinued) assignment location  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT fsm.msny_id  
            ,'Missionary assigned to discontinued assignment location ' ||  
             fsm.comp_asgloc_nm || ' (asgloc_id=' || fsm.comp_asgloc_id || ').'  
            ,'NAPI_ASSIGNMENT_MSTR'  
            ,'CALL_ALOC_ID'  
        FROM fin_sweep_msny fsm  
        JOIN assignment_locations al  
          ON al.asgloc_id = fsm.comp_asgloc_id  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND al.asglocsta_id = 5;  
         
   -- Assignment location not found in CDOL anymore  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Assignment location not found in CDOL (orgnbr='||al.asgloc_col_orgnbr||', name='||al.name||')'  
            ,'ASSIGNMENT_LOCATIONS'  
            ,'ASGLOC_COL_ORGNBR'  
        FROM fin_sweep_msny_gt gt  
             JOIN current_perm_assignment_vw ca  
                  ON ca.call_emem_id = gt.enamem_id  
             JOIN assignment_locations al  
                  ON al.asgloc_id = ca.call_aloc_id  
             LEFT JOIN fin_organization_mv mcom  
                  ON mcom.orgnbr = al.asgloc_col_orgnbr  
       WHERE mcom.orgid IS NULL;  
                           
    -- Mission Country not in our Poldiv to Country translator  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm)  
      SELECT gt.msny_id  
            ,'Mission country ('||mcom.primarylocpoldivid||') missing from POLDIV_TO_COUNTRY_MV.'  
            ,'POLDIV_TO_COUNTRY_MV'  
        FROM fin_sweep_msny_gt gt  
             JOIN current_perm_assignment_vw ca  
                  ON ca.call_emem_id = gt.enamem_id  
             JOIN assignment_locations al  
                  ON al.asgloc_id = ca.call_aloc_id  
             JOIN fin_organization_mv mcom  
                  ON mcom.orgnbr = al.asgloc_col_orgnbr  
             LEFT JOIN poldiv_to_country_mv mpc  
                  ON mpc.primary_poldivid = mcom.primarylocpoldivid  
       WHERE mpc.primary_poldivid IS NULL;  
     
   -- Mission area not found in CDOL  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT fsm.msny_id  
            ,'Mission area is blank in CDOL for mission (orgnbr='||fsm.asgloc_orgnbr||').'  
            ,'COL_ORGANIZATION_MV'  
            ,'AREAORGNBR'  
        FROM fin_sweep_msny fsm  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND fsm.asgloc_orgnbr IS NOT NULL  
         AND fsm.asgloc_area_orgnbr IS NULL;  
        
   -- Funding unit Country not in our Poldiv to Country translator  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT gt.msny_id  
            ,'Cannot get country from funding unit ('||fmf.fund_unit_id||').'  
            ,'COL_ORGANIZATION_MV'  
            ,'PRIMARYLOCPOLDIVID'  
        FROM fin_sweep_msny_gt gt  
             JOIN fin_msny_funding fmf  
                  ON fmf.enamem_id = gt.enamem_id  
             JOIN fin_organization_mv fcom  
                  ON fcom.orgnbr = fmf.fund_unit_id  
             LEFT JOIN poldiv_to_country_mv fpc  
                  ON fpc.primary_poldivid = fcom.primarylocpoldivid  
       WHERE fmf.fund_type_id IN (1, 2, 4, 5)  
         AND fpc.primary_poldivid IS NULL;  
           
   -- Funding unit Country not in our Poldiv to Country translator  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm)  
      SELECT gt.msny_id  
            ,'Funding unit country (primary_poldivid='||fcom.primarylocpoldivid||') missing from POLDIV_TO_COUNTRY_MV.'  
            ,'POLDIV_TO_COUNTRY_MV'  
        FROM fin_sweep_msny_gt gt  
             JOIN fin_msny_funding fmf  
                  ON fmf.enamem_id = gt.enamem_id  
             JOIN fin_organization_mv fcom  
                  ON fcom.orgnbr = fmf.fund_unit_id  
             LEFT JOIN poldiv_to_country_mv fpc  
                  ON fpc.primary_poldivid = fcom.primarylocpoldivid  
       WHERE fmf.fund_type_id IN (1, 2, 4, 5)  
         AND fcom.primarylocpoldivid IS NOT NULL  
         AND fpc.primary_poldivid IS NULL;  
  
   -- Mission country not in our poldiv financial attributes tables  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm)  
      SELECT gt.msny_id  
            ,'Mission country '||mpc.country_poldivid||' missing from FIN_POLDIVS.'  
            ,'FIN_POLDIVS'  
        FROM fin_sweep_msny_gt gt  
             JOIN current_perm_assignment_vw ca  
                  ON ca.call_emem_id = gt.enamem_id  
             JOIN assignment_locations al  
                  ON al.asgloc_id = ca.call_aloc_id  
             JOIN fin_organization_mv mcom  
                  ON mcom.orgnbr = al.asgloc_col_orgnbr  
             JOIN poldiv_to_country_mv mpc -- join for mission country  
                  ON mpc.primary_poldivid = mcom.primarylocpoldivid   
             LEFT JOIN fin_poldivs mfp -- join for mission currency  
                  ON mfp.poldivid = mpc.country_poldivid  
       WHERE mfp.poldivid IS NULL;  
         
   -- Downstream processing needs stake of funding unit  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt  
      ,table_nm  
      ,column_nm)  
      SELECT fsm.msny_id  
            ,'Cannot get stake from funding unit ('||fsm.fund_unit_id||'). Funding unit is '||com.orgstatuscodedesc||'.'  
            ,'COL_ORGANIZATION_MV'  
            ,'PARENTORGNBR'  
        FROM fin_sweep_msny fsm  
        JOIN fin_organization_mv com ON com.orgnbr = fsm.fund_unit_id  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND fsm.fund_type_id IN (1, 2, 4, 5)  
         AND fsm.fund_unit_stake_id IS NULL;  
      
   -- Downstream processing needs country of funding unit  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt)  
      SELECT fsm.msny_id  
            ,'Cannot get country from funding unit ('||fsm.fund_unit_id||')'||CHR(10)||  
             'col_organization_mv.primarylocpoldivid='||fcom.primarylocpoldivid||CHR(10)||  
             'poldiv_to_country_mv.country_poldivid='||ptc.country_poldivid||CHR(10)||  
             'fin_poldivs.poldivid='||ffp.poldivid  
        FROM fin_sweep_msny fsm  
        LEFT JOIN fin_organization_mv fcom  
             ON fcom.orgnbr = fsm.fund_unit_id  
        LEFT JOIN poldiv_to_country_mv ptc  
             ON ptc.primary_poldivid = fcom.primarylocpoldivid  
        LEFT JOIN fin_poldivs ffp  
             ON ffp.poldivid = ptc.country_poldivid  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND fsm.fund_type_id IN (1, 2, 4, 5)  
         AND fsm.fund_unit_country_id IS NULL;  
           
   -- Currency from funding unit  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt)  
      SELECT fsm.msny_id  
            ,'Cannot get currency from funding unit ('||fsm.fund_unit_id||')'||CHR(10)||  
             'col_organization_mv.primarylocpoldivid='||fcom.primarylocpoldivid||CHR(10)||  
             'poldiv_to_country_mv.country_poldivid='||ptc.country_poldivid||CHR(10)||  
             'fin_poldivs.finpol_iso_currency_code='||ffp.finpol_iso_currency_code  
        FROM fin_sweep_msny fsm  
        LEFT JOIN fin_organization_mv fcom  
             ON fcom.orgnbr = fsm.fund_unit_id  
        LEFT JOIN poldiv_to_country_mv ptc  
             ON ptc.primary_poldivid = fcom.primarylocpoldivid  
        LEFT JOIN fin_poldivs ffp  
             ON ffp.poldivid = ptc.country_poldivid  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND fsm.fund_type_id IN (1, 2, 4, 5)  
         AND fsm.fund_unit_currency_cd IS NULL;  
  
   -- Currency from mission (should be non-zero rarely or never)  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt)  
      SELECT fsm.msny_id  
            ,'Cannot get currency from mission (orgnbr='||fsm.asgloc_orgnbr||')'  
        FROM fin_sweep_msny fsm  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND fsm.asgloc_orgnbr IS NOT NULL  
         AND fsm.msn_currency_cd IS NULL;  
  
   -- Country from mission (should be non-zero rarely or never)  
   l_marker := l_marker + 1;  
   logs.dbg('Sweep Error Check #'||l_marker);  
   INSERT INTO fin_sweep_msny_err  
      (msny_id  
      ,err_txt)  
      SELECT fsm.msny_id  
            ,'Cannot get country from mission (orgnbr='||fsm.asgloc_orgnbr||')'  
        FROM fin_sweep_msny fsm  
       WHERE fsm.sweep_dt = l_sweep_dt  
         AND fsm.asgloc_orgnbr IS NOT NULL  
         AND fsm.asgloc_country_id IS NULL;  
               
END run_error_check;  
  
--------------------------------------------------------------------------------  
--                        PUBLIC FUNCTIONS AND PROCEDURES  
--------------------------------------------------------------------------------  
  
--------------------------------------------------------------------------------  
PROCEDURE snap_exchange_rate_map  
(  
   i_snap_dt IN DATE DEFAULT SYSDATE  
)  
IS
   CURSOR cur_diffs(i_date IN DATE) IS
      SELECT fer.fin_exch_rate_id
           , fer.from_cur, fer.exch_rate, fer.eff_dt
           , psr.exch_rate AS new_exch_rate, psr.eff_dt AS new_eff_dt
        FROM fin_exch_rate fer
        -- inline view to get latest rates from PeopleSoft copy
        JOIN (SELECT from_cur, to_cur, cur_exchng_rt AS exch_rate, effdt AS eff_dt
                FROM (SELECT from_cur, to_cur, cur_exchng_rt, effdt
                           , ROW_NUMBER() OVER (PARTITION BY from_cur, to_cur ORDER BY effdt DESC) rn -- determine latest rate   
                        FROM ps_cur_rt_tbl_mv   
                       WHERE effdt <= i_date -- filter out PS rates newer than the 1st of the month  
                         AND cur_rt_type = 'CRRNT' -- only Current exchange rates  
                         AND to_cur = 'USD')   
               WHERE rn = 1
             ) psr ON psr.from_cur = fer.from_cur
      WHERE fer.exch_rate <> psr.exch_rate
        AND psr.eff_dt > fer.eff_dt;

   -- although meant to only be called by the monthly sweep job, if called manually  
   -- at other times in the month, we force it to the month start anyway  
   l_snap_dt DATE := TRUNC(NVL(i_snap_dt,SYSDATE),'Month');  
   l_rows INTEGER := 0;
BEGIN  
     
   logs.dbg('Updating copy of latest PeopleSoft/CIFS exchange rates for USD conversions...');
   
   FOR lr IN cur_diffs(l_snap_dt) LOOP
      UPDATE fin_exch_rate
         SET exch_rate = lr.new_exch_rate
           , eff_dt = lr.new_eff_dt
       WHERE fin_exch_rate_id = lr.fin_exch_rate_id;
       l_rows := l_rows + 1;
   END LOOP;
         
   logs.dbg('Updated '||l_rows||' PS exchange rate mappings...');  
      
   COMMIT;  
     
END snap_exchange_rate_map;  
  
--------------------------------------------------------------------------------  
PROCEDURE run_sweep  
(  
   i_sweep_dt IN DATE DEFAULT SYSDATE  
) IS  
   -- although they can run the sweep at any time, we force it to the month start  
   l_sweep_dt DATE := TRUNC(NVL(i_sweep_dt,SYSDATE),'Month');  
   l_proc_hist_id fin_proc_hist.proc_hist_id%TYPE := 0;  
BEGIN  
  
   -- Front end should be setting the client id to the user's login upon connection to the DB pool.  
   -- When the front end logs into the database, it is through a PTC account which  
   -- has an AFTER LOGON trigger that sets the app code.  
   -- The monthly job that calls this routine also sets the client ID and app code.  
   -- If you are attempting to call this routine in anonymous PL/SQL block, either do so  
   -- by logging into the PTC account first, or set the client ID and app code   
   -- manually. See the "what" text for the Monthly Sweep job as an example.  
     
   -- Validate parameter assumptions  
   excp.assert(l_sweep_dt = TRUNC(SYSDATE,'Month'),'Unsupported use. It is currently not allowed to run a sweep for a prior or future month.');  
  
   -- Ensure sweep for this month isn't already finalized  
   excp.assert(sweep_finalized(l_sweep_dt) = FALSE,'Sweep for this month has already been finalized. Cannot run again.');  
     
   logs.info('BEGIN');  
     
   -- Add row to process history table  
   tag_proc_hist(l_sweep_dt, gc_start, l_proc_hist_id, 'RUNNING', 'Sweep Started.');  
     
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Cleaning up prior run for '||TO_CHAR(l_sweep_dt,'YYYYMonDD'));  
   clean_prior_sweep(l_sweep_dt);  
     
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Load temp table with sweep set');  
   load_sweep_set(l_sweep_dt);  
  
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Adjusting fund_type_id');  
   -- Keep fund_type_id on fin_msny_funding updated with changes to countries and units (fin_poldivs and fin_church_units)  
   adj_msny_fund_type_id();  
     
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Adding detail data to Sweep Missionary table');  
   ins_sweep_msny(l_sweep_dt);  
  
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Running error diagnostics');  
   run_error_check(l_sweep_dt);  
     
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Inserting into Sweep Tally table');  
   ins_sweep_tally(l_sweep_dt);  
  
   COMMIT; -- we decided to let the driver be in charge of the commit, instead of the caller.  
     
   -- Cap off row in process history table  
   tag_proc_hist(l_sweep_dt, gc_end, l_proc_hist_id, 'SUCCESS', 'Sweep Ended.');  
     
   logs.info('END');  
  
EXCEPTION  
   WHEN OTHERS THEN  
      tag_proc_hist(l_sweep_dt, gc_end, l_proc_hist_id, 'ERROR', SQLERRM);  
      RAISE;     
END run_sweep;  
  
--------------------------------------------------------------------------------  
PROCEDURE finalize_sweep  
(  
   i_proc_hist_id IN fin_proc_hist.proc_hist_id%TYPE  
) IS  
   l_sweep_dt DATE;  
   l_proc_hist_id fin_proc_hist.proc_hist_id%TYPE;  
     
   FUNCTION get_sweep_dt RETURN DATE  
   IS  
      l_sweep_dt DATE;  
   BEGIN  
      SELECT proc_dt  
        INTO l_sweep_dt  
        FROM fin_proc_hist  
       WHERE proc_hist_id = i_proc_hist_id;  
      RETURN l_sweep_dt;    
   EXCEPTION  
      WHEN NO_DATA_FOUND THEN  
        logs.msg(msgs.ERROR_MSG_CD,cnst.ERROR,'proc_hist_id: '||i_proc_hist_id||  
                ' is not a valid value.',TRUE);  
   END get_sweep_dt;  
BEGIN  
  
   -- If job did not set, then make sure a client ID is set. Front end should be  
   -- setting the client id to the user's login upon connection to the DB pool.  
   IF (env.get_client_id IS NULL) THEN  
      env.init_client_ctx(i_client_id => USER, i_app_cd => 'RCRD');  
   END IF;  
  
   -- Validate parameter assumptions  
   excp.assert(i_proc_hist_id IS NOT NULL,  
      'The ID of latest successful sweep record in FIN_PROC_HIST must be supplied.');  
   l_sweep_dt := get_sweep_dt;  
  
   -- Validate parameter assumptions  
   excp.assert(l_sweep_dt = TRUNC(SYSDATE,'Month'),  
      'Unsupported use. Cannot finalize a sweep for a prior or future month.');  
  
   l_proc_hist_id := i_proc_hist_id;  
        
   -- Ensure sweep for this month isn't already finalized  
   excp.assert(sweep_finalized(l_sweep_dt) = FALSE,'Sweep for '||  
      TO_CHAR(l_sweep_dt,'Month YYYY')||' has already been finalized. Cannot finalize again.');  
     
   logs.info('BEGIN');  
     
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Removing prior payments for this month');  
   clean_prior_payments(l_sweep_dt);  
     
   tag_proc_hist(l_sweep_dt, gc_update, l_proc_hist_id, NULL, 'Inserting to Payment History table');  
   ins_pay_hist(l_sweep_dt);  
     
   COMMIT; -- we decided to let the driver be in charge of the commit, instead of the caller.  
    
   -- Cap off row in process history table  
   tag_proc_hist(l_sweep_dt, gc_end, l_proc_hist_id, 'FINALIZED', 'Sweep Finalized.');  
     
   logs.info('END');  
  
   env.reset_client_ctx; 

EXCEPTION  
   WHEN OTHERS THEN  
      tag_proc_hist(l_sweep_dt, gc_end, l_proc_hist_id, 'ERROR', SQLERRM);  
      RAISE;     
END finalize_sweep;  

--------------------------------------------------------------------------------
PROCEDURE weekly_tally(p_run_date DATE DEFAULT SYSDATE ) AS
 
  lv_run_date  DATE;  
 
BEGIN  
 
   lv_run_date := TRUNC(p_run_date);  
 
   EXECUTE IMMEDIATE 'TRUNCATE TABLE msny_week_tally_cand_gt';  
  
   /*** list of missionaries to process ***/  
   INSERT INTO msny_week_tally_cand_gt   
      SELECT m.id  
            ,m.enamem_id  
            ,m.legacy_miss_id  
            ,m.mrn  
            ,mn.first_nm||DECODE(em.misstyp_id,3,' and '||mnw.first_nm,NULL)
            ,mn.last_nm
            ,DECODE(em.misstyp_id,3,NULL,mn.middle_nm)
            ,em.term_months pmts_to_send  
            ,SUM(DECODE(fph.dr_cr_code, 'DR', 1, 'CR', -1)) pmts_sent  
        FROM enabled_members em
        JOIN missionaries m ON m.legacy_miss_id = em.legacy_miss_id -- only joins husband for couples
        JOIN current_perm_assignment_vw cpa ON cpa.call_emem_id = em.enamem_id AND cpa.early_release_flag <> 'Y' -- RMH Records-4102 Filter out early released missionaries
        JOIN msny_name mn ON mn.legacy_miss_id = em.legacy_miss_id AND mn.name_type_id = 10 -- Candidate name 
        LEFT JOIN msny_name mnw ON mnw.legacy_miss_id = em.legacy_spouse_miss_id AND mnw.name_type_id = 10 -- Wife's Candidate name 
        LEFT JOIN fin_pay_hist fph ON fph.legacy_miss_id = m.legacy_miss_id
       WHERE em.procstat_id IN (11, 12, 13, 14)  
         AND em.enamem_release_date >= lv_run_date  
       GROUP BY m.id  
               ,m.enamem_id  
               ,m.legacy_miss_id  
               ,m.mrn  
               ,mn.first_nm||DECODE(em.misstyp_id,3,' and '||mnw.first_nm,NULL)
               ,mn.last_nm
               ,DECODE(em.misstyp_id,3,NULL,mn.middle_nm)
               ,em.term_months; 
   COMMIT;  
     
   /**** bring together temp msny, fund, and serving info ****/  
     
   EXECUTE IMMEDIATE 'TRUNCATE TABLE msny_week_tally_gt';  
     
   INSERT INTO msny_week_tally_gt   
      ( sweep_msny_id  
      , sweep_dt   
      ,created_dt   
      ,msny_id   
      ,legacy_miss_id   
      ,enamem_id  
      ,citizenship_id   
      ,procstat_id   
      ,msny_nm   
      ,misstyp_id   
      ,term_months   
      ,live_at_home_yn   
      ,mtc_dt   
      ,arriv_dt   
      ,release_dt   
      ,anniv_dt   
      ,mtc_id   
      ,mtc_cd   
      ,mtc_nm   
      ,call_type_id   
      ,comp_id   
      ,comp_nm   
      ,comp_asgloc_orgnbr   
      ,comp_asgloc_id   
      ,comp_asgloc_nm   
      ,comp_asgloc_orgtype_cd   
      ,comp_asgloc_orgtype_nm   
      ,comp_asgloc_country_id   
      ,comp_asgloc_country_nm   
      ,comp_asgloc_sens_flg           
      ,asgloc_orgnbr   
      ,asgloc_id   
      ,asgloc_nm   
      ,asgloc_orgtype_cd   
      ,asgloc_orgtype_nm   
      ,asgloc_area_orgnbr   
      ,asgloc_area_nm   
      ,asgloc_country_id   
      ,asgloc_country_nm   
      ,asgloc_sens_flg               
      ,within_msn_orgnbr   
      ,within_msn_asgloc_id   
      ,within_msn_nm   
      ,within_msn_sens_flg           
      ,eccl_org_orgnbr   
      ,eccl_org_nm   
      ,eccl_org_sens_flg             
      ,temp_comp_id   
      ,temp_comp_nm   
      ,temp_comp_asgloc_orgnbr   
      ,temp_comp_asgloc_id   
      ,temp_comp_asgloc_nm   
      ,temp_comp_asgloc_sens_flg     
      ,temp_asgloc_orgnbr   
      ,temp_asgloc_id   
      ,temp_asgloc_nm   
      ,temp_asgloc_sens_flg          
      ,temp_within_msn_orgnbr   
      ,temp_within_msn_asgloc_id   
      ,temp_within_msn_nm   
      ,temp_within_msn_sens_flg      
      ,temp_eccl_org_orgnbr   
      ,temp_eccl_org_nm   
      ,temp_eccl_org_sens_flg        
      ,fund_type_id   
      ,last_pmt_dt   
      ,blbl_month_yn   
      ,crnt_pmt_num   
      ,total_pmts   
      ,fund_unit_id   
      ,fund_unit_nm   
      ,fund_unit_sens_flg            
      ,fund_unit_stake_id   
      ,fund_unit_stake_nm   
      ,fund_unit_stake_sens_flg      
      ,fund_unit_area_orgnbr   
      ,fund_unit_area_nm   
      ,fund_unit_country_id   
      ,fund_unit_country_nm   
      ,fund_unit_currency_used   
      ,fund_unit_currency_cd   
      ,fund_unit_currency_nm   
      ,fund_unit_exchange_rt   
      ,fund_unit_equal_amt   
      ,fund_unit_commit_amt   
      ,fund_unit_local_amt   
      ,fund_unit_us_amt   
      ,msn_currency_used   
      ,msn_currency_cd   
      ,msn_currency_nm   
      ,msn_local_amt   
      ,msn_us_amt   
      ,msn_exchange_rt   
      ,msn_facs_num   
      )   
   SELECT    
          tallywkly_seq.nextval  
          ,TRUNC(lv_run_date-7) AS sweep_dt   
          ,TRUNC(lv_run_date) AS created_dt   
          ,gt.msny_id   
          ,gt.legacy_miss_id   
          ,gt.enamem_id  
          ,em.enamem_homecountry_id  
          ,em.procstat_id   
          ,RTRIM(gt.last_nm || ', ' || gt.first_nm || ' ' || gt.middle_nm) AS msny_nm   
          ,em.misstyp_id   
          ,em.term_months   
          ,DECODE(alc.comp_housing_id,3,'Y','N') AS live_at_home_yn -- fixed during Sr. Exp work   
          ,em.enamem_mission_start_date AS mtc_dt   
          ,ca.asgmnt_start_date AS arriv_dt   
          ,em.enamem_release_date AS release_dt   
          ,em.enamem_anniversary_date AS anniv_dt   
          ,ca.call_mtcs_id AS mtc_id   
          ,ctom.orgshortname AS mtc_cd   
          ,tf.name AS mtc_nm   
          ,ca.call_type AS call_type_id   
          ,alc.comp_id AS comp_id   
          ,alc.comp_description AS comp_nm   
          ,cfom.orgnbr AS comp_asgloc_orgnbr   
          ,alc.asgloc_id AS comp_asgloc_id   
          ,cal.name AS comp_asgloc_nm   
          ,cfom.orgtypecode AS comp_asgloc_orgtype_cd   
          ,cfom.orgtypeshortdesctxt AS comp_asgloc_orgtype_nm   
          ,cpc.country_poldivid AS comp_asgloc_country_id   
          ,cpc.country_longname AS comp_asgloc_country_nm   
          ,cfom.orgsensitivitycode AS comp_asgloc_sens_flg   
          ,mfom.orgnbr AS asgloc_orgnbr   
          ,ca.call_aloc_id AS asgloc_id   
          ,mama.asgloc_name AS asgloc_nm   
          ,mfom.orgtypecode AS asgloc_orgtype_cd   
          ,mfom.orgtypeshortdesctxt AS asgloc_orgtype_nm   
          ,mama.area_orgnbr AS asgloc_area_orgnbr   
          ,mama.area_name AS asgloc_area_nm   
          ,mpc.country_poldivid AS asgloc_country_id   
          ,mpc.country_longname AS asgloc_country_nm   
          ,mfom.orgsensitivitycode AS asgloc_sens_flg   
          ,mama.wm_orgnbr AS within_msn_orgnbr   
          ,wmfom.asgloc_id AS within_msn_asgloc_id   
          ,mama.wm_orglongname AS within_msn_nm   
          ,wmfom.orgsensitivitycode AS within_msn_sens_flg   
          ,mama.eccl_orgnbr AS eccl_org_orgnbr   
          ,mama.eccl_orglongname AS eccl_org_nm   
          ,efom.orgsensitivitycode AS eccl_org_sens_flg   
          ,cta.call_comp_id AS temp_comp_id   
          ,talc.comp_description AS temp_comp_nm   
          ,tcfom.orgnbr AS temp_comp_asgloc_orgnbr   
          ,talc.asgloc_id AS temp_comp_asgloc_id   
          ,tcal.name AS temp_comp_asgloc_nm   
          ,NVL(tcfom.orgsensitivitycode,0) AS temp_comp_asgloc_sens_flg   
          ,tmfom.orgnbr AS temp_asgloc_orgnbr   
          ,cta.call_aloc_id AS temp_asgloc_id   
          ,tama.asgloc_name AS temp_asgloc_nm   
          ,NVL(tmfom.orgsensitivitycode,0) AS temp_asgloc_sens_flg   
          ,tama.wm_orgnbr AS temp_within_msn_orgnbr   
          ,twmfom.asgloc_id AS temp_within_msn_asgloc_id   
          ,tama.wm_orglongname AS temp_within_msn_nm   
          ,NVL(twmfom.orgsensitivitycode,0) AS temp_within_msn_sens_flg   
          ,tama.eccl_orgnbr AS temp_eccl_org_orgnbr   
          ,tama.eccl_orglongname AS temp_eccl_org_nm   
          ,NVL(tefom.orgsensitivitycode,0) AS temp_eccl_org_sens_flg   
          ,fmf.fund_type_id   
          ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN lpm.last_pay_dt ELSE NULL END AS last_pmt_dt   
          ,CASE   
            WHEN (lpm.last_pay_dt is NULL OR fmf.fund_type_id IS NULL)
               THEN 'N' -- not enough to go on
            WHEN (em.procstat_id = 11 AND TRUNC(em.enamem_mission_start_date, 'Month') > TRUNC(lv_run_date))
               THEN 'N' -- prefield that won't be in the MTC this month
            WHEN fmf.fund_type_id = 3 OR (MONTHS_BETWEEN(TRUNC(lpm.last_pay_dt,'Month'),lv_run_date) < 0)
               THEN 'N' -- unsupported or those whose last payment month has passed
             ELSE 'Y'   
           END AS blbl_month_yn   
          ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN NVL(gt.pmts_sent,0)+1 ELSE NULL END AS crnt_pmt_num   
          ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN NVL(gt.pmts_to_send,0) ELSE NULL END AS total_pmts   
          ,fmf.fund_unit_id AS fund_unit_id   
          ,ffom.orglongname AS fund_unit_nm   
          ,NVL(ffom.orgsensitivitycode,0) AS fund_unit_sens_flg   
          ,ffom.parentorgnbr AS fund_unit_stake_orgnbr   
          ,fstake.orglongname AS fund_unit_stake_nm   
          ,NVL(fstake.orgsensitivitycode,0) AS fund_unit_stake_sens_flg   
          ,ffom.areaorgnbr AS fund_unit_area_orgnbr   
          ,farea.orglongname AS fund_unit_area_nm   
          ,ffpc.country_poldivid AS fund_unit_country_id   
          ,ffpc.country_longname AS fund_unit_country_nm   
          ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN ffp.cfis_local_or_us_flag ELSE NULL END AS fund_unit_currency_used   
          ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN fcurr.iso_currency_code ELSE NULL END AS fund_unit_currency_cd   
          ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN fcurr.currency_name ELSE NULL END AS fund_unit_currency_nm   
          ,CASE WHEN fmf.fund_type_id IN (1,2,4,5) THEN fpcr.exch_rate ELSE NULL END AS fund_unit_exchange_rt   
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN ffp.finpol_equalized_pay_amt 
              WHEN fmf.fund_type_id IN (4,5) THEN alc.housing_exp
              ELSE NULL
           END AS fund_unit_equal_amt  
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN
                 -- add up commitment sources as reported by junior missionary  
                 NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0)
              WHEN fmf.fund_type_id IN (4,5) THEN
                 -- get housing commitment amount
                 fmf.housing_cmt_amt
          END AS fund_unit_commit_amt   
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN
                  -- Local Amount Due from Funding Unit = commitments if not equalized, otherwise the equalized amount
                  -- This formula for local and US amounts came from Legacy Pro*C file sweeptally.pc  
                  CASE  
                   -- check unit override table for equalization flag, before using eq flag for country  
                   WHEN NVL(fuo.equalized_yn, ffp.finpol_equalized_yn) = 'Y' THEN  
                      ffp.finpol_equalized_pay_amt  
                   ELSE   
                      LEAST(NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0),  
                            CASE  
                               WHEN (ffp.finpol_equalized_pay_amt IS NULL OR ffp.finpol_equalized_pay_amt = 0) THEN   
                                  NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0)  
                               ELSE  
                                  ffp.finpol_equalized_pay_amt  
                            END  
                      )  
                  END  
               WHEN fmf.fund_type_id = 4 THEN
                  CASE WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y' AND fmf.housing_exp_amt > 0) THEN
                          fmf.housing_exp_amt -- missionary housing expense is stored in local currency of funding unit
                       ELSE 0
                  END
               WHEN fmf.fund_type_id = 5 THEN
                  CASE
                     WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y') THEN
                        -- Compare housing commitment to housing expense. Take the lesser.
                        CASE WHEN fmf.housing_cmt_amt <= fmf.housing_exp_amt THEN
                          fmf.housing_cmt_amt
                             WHEN fmf.housing_cmt_amt > fmf.housing_exp_amt THEN
                                fmf.housing_exp_amt
                        END
                     ELSE 0
                  END
               ELSE 0
          END AS fund_unit_local_amt  
          -- USD amt due same as local, but multiplied by USD exchange rate  
         ,CASE WHEN fmf.fund_type_id IN (1,2) THEN
                  -- USD Amount Due from Funding Unit = commitments if not equalized, otherwise the equalized amount
                  -- This formula for local and US amounts came from Legacy Pro*C file sweeptally.pc  
                  ROUND(  
                     CASE  
                      -- check unit override table for equalization flag, before using eq flag for country  
                      WHEN NVL(fuo.equalized_yn, ffp.finpol_equalized_yn) = 'Y' THEN  
                         ffp.finpol_equalized_pay_amt  
                      ELSE   
                         LEAST(NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0),  
                               CASE  
                                  WHEN (ffp.finpol_equalized_pay_amt IS NULL OR ffp.finpol_equalized_pay_amt = 0) THEN   
                                     NVL(fmf.fin_self,0) + NVL(fmf.fin_family,0) + NVL(fmf.fin_ward,0) + NVL(fmf.fin_other,0)  
                                  ELSE  
                                     ffp.finpol_equalized_pay_amt  
                               END  
                         )  
                     END * fpcr.exch_rate 
                  )
               WHEN fmf.fund_type_id = 4 THEN
                  CASE WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y' AND fmf.housing_exp_amt > 0) THEN
                          ROUND(fmf.housing_exp_amt * fpcr.exch_rate)-- missionary housing expense in terms of USD
                       ELSE 0
                  END
               WHEN fmf.fund_type_id = 5 THEN
                  CASE
                     WHEN (alc.sweep_yn = 'Y' AND fmf.housing_cap_ppt_yn = 'Y') THEN
                        -- Compare commitment to housing expense and take the lesser
                        ROUND(
                           CASE WHEN fmf.housing_cmt_amt <= fmf.housing_exp_amt THEN
                                   fmf.housing_cmt_amt
                                WHEN fmf.housing_cmt_amt > fmf.housing_exp_amt THEN
                             fmf.housing_exp_amt
                           END * fpcr.exch_rate
                        )
                     ELSE 0
                  END
               ELSE 0
           END AS fund_unit_us_amt  
          ,mfp.cfis_local_or_us_flag AS msn_currency_used   
          ,mcurr.iso_currency_code AS msn_currency_cd   
          ,mcurr.currency_name AS msn_currency_nm   
           -- this formula for local and US amounts came from Legacy Pro*C file sweeptally.pc   
          ,CASE WHEN mfp.cfis_local_or_us_flag = 'L' THEN mfa.base_amt ELSE ROUND(mfa.base_amt / mpcr.exch_rate) END AS msn_local_amt   
          ,CASE WHEN mfp.cfis_local_or_us_flag = 'L' THEN ROUND(mfa.base_amt * mpcr.exch_rate) ELSE mfa.base_amt END AS msn_us_amt   
          ,mpcr.exch_rate AS msn_exchange_rt   
          ,mfa.gl_unit_num AS msn_facs_num   
        FROM msny_week_tally_cand_gt gt -- this is the filter for what gets swept   
             JOIN enabled_members em ON em.enamem_id = gt.enamem_id   
             JOIN current_perm_assignment_vw ca ON ca.call_emem_id = gt.enamem_id   
             JOIN fin_msny_funding fmf ON fmf.enamem_id = gt.enamem_id   
             JOIN (   
                    SELECT g.msny_id,    
                           CASE   
                              -- How many days beyond planned release is the actual release?   
                              WHEN (e.enamem_release_date IS NULL OR e.enamem_anniversary_date IS NULL) THEN   
                                NULL -- return empty value to be picked up by error module   
                              WHEN (e.enamem_release_date - e.enamem_anniversary_date) > 14 THEN   
                              ADD_MONTHS(e.enamem_anniversary_date, 
                                         (TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))
                                        +(CASE
                                             WHEN e.enamem_release_date - (ADD_MONTHS(e.enamem_anniversary_date,TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))) > 14
                                                THEN 0
                                             ELSE -1
                                          END)
                                        )  
                              WHEN (e.enamem_release_date - e.enamem_anniversary_date) < -14 THEN   
                              ADD_MONTHS(e.enamem_anniversary_date, 
                                         (TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))
                                        -(CASE
                                             WHEN e.enamem_release_date - (ADD_MONTHS(e.enamem_anniversary_date,TRUNC(MONTHS_BETWEEN(e.enamem_release_date,e.enamem_anniversary_date)))) < -14
                                                THEN 2
                                             ELSE 1
                                          END)
                                        )  
                              ELSE   
                                ADD_MONTHS(e.enamem_anniversary_date,-1)   
                           END AS last_pay_dt   
                 FROM fin_sweep_msny_gt g  
                 JOIN enabled_members e ON e.enamem_id = g.enamem_id  
                  ) lpm   
                  ON lpm.msny_id = gt.msny_id   
             JOIN asg_loc_components alc ON alc.comp_id = ca.call_comp_id   
             -- component-level asgloc info   
             JOIN fin_organization_mv cfom  ON cfom.asgloc_id = alc.asgloc_id   
             JOIN assignment_locations cal ON cal.asgloc_id = alc.asgloc_id   
             JOIN poldiv_to_country_mv cpc ON cpc.primary_poldivid = cfom.primarylocpoldivid   
             -- "parent" asgloc info   
             JOIN fin_organization_mv mfom ON mfom.asgloc_id = ca.call_aloc_id   
             JOIN asgloc_msn_area_mv mama ON mama.asgloc_orgnbr = mfom.orgnbr   
             JOIN poldiv_to_country_mv mpc ON mpc.primary_poldivid = mfom.primarylocpoldivid   
             -- "within mission" info   
             JOIN fin_organization_mv wmfom ON wmfom.orgnbr = mama.wm_orgnbr   
             -- ecclesiastical info   
             JOIN fin_organization_mv efom ON efom.orgnbr = mama.eccl_orgnbr   
             -- joins to get financial info for the mission    
             JOIN fin_poldivs mfp ON mfp.poldivid = mpc.country_poldivid   
             JOIN mdm_currency mcurr ON mcurr.iso_currency_code = mfp.finpol_iso_currency_code   
             JOIN fin_exch_rate mpcr ON mpcr.from_cur = mcurr.iso_currency_code   
             JOIN fin_asglocs mfa ON mfa.asgloc_id = mfom.asgloc_id   
             -- start of outer joins ---------------------------------------------
             LEFT JOIN training_facilities tf ON tf.trnfac_id = ca.call_mtcs_id -- many MTC IDs are null, some 0   
             LEFT JOIN col_training_orgs_mv ctom ON ctom.orgnbr = tf.col_orgnbr -- must left join as well if MTC was null   
             LEFT JOIN fin_organization_mv ffom ON ffom.orgnbr = fmf.fund_unit_id -- some test missionaries won't have entries in fin_organization_mv  
             LEFT JOIN fin_organization_mv fstake ON fstake.orgnbr = ffom.parentorgnbr -- need long name of stake   
             LEFT JOIN fin_organization_mv farea ON farea.orgnbr = ffom.areaorgnbr -- need long name of area   
             LEFT JOIN poldiv_to_country_mv fpc ON fpc.primary_poldivid = ffom.primarylocpoldivid   
             LEFT JOIN fin_unit_override fuo  
                  ON fuo.orgnbr = CASE  
                                     WHEN fmf.allow_override_yn = 'Y' THEN ffom.orgnbr   
                                     ELSE NULL -- remove ability to join to override table   
                                  END   
             LEFT JOIN fin_poldivs ffp -- join for funding unit country financial attributes   
                  ON ffp.poldivid = CASE    
                                       WHEN fuo.orgnbr IS NULL THEN fpc.country_poldivid    
                                       ELSE fuo.poldivid_of_equalized_amt    
                                    END   
             -- Now that fin_poldivs is joined, perhaps on redirected country from poldivid_of_equalized_amt,   
             -- re-join to poldiv_to_country_mv to get country info.   
             LEFT JOIN poldiv_to_country_mv ffpc ON ffpc.primary_poldivid = ffp.poldivid   
             LEFT JOIN mdm_currency fcurr ON fcurr.iso_currency_code = ffp.finpol_iso_currency_code   
             LEFT JOIN fin_exch_rate fpcr ON fpcr.from_cur = fcurr.iso_currency_code   
             -- Plethora of outer joins to get possible temp assignment info    
             LEFT JOIN (   
                -- current_temp_assignment_vw didn't work since it internally used SYSDATE to filter. We   
                -- need the filter to use the sweep date, weekly   
                SELECT call_id, call_aloc_id, call_comp_id, call_emem_id, asgmnt_start_date, asgmnt_end_date   
                 FROM napi_assignment_mstr
                WHERE call_type IN (2,3)
                  AND asgmnt_start_date <= lv_run_date  
                  AND (asgmnt_end_date > lv_run_date OR asgmnt_end_date IS NULL)   
                  AND status <> 7   
             ) cta ON cta.call_emem_id = em.enamem_id   
             LEFT JOIN asg_loc_components talc ON talc.comp_id = cta.call_comp_id   
             LEFT JOIN fin_organization_mv tcfom ON tcfom.asgloc_id = talc.asgloc_id   
             LEFT JOIN assignment_locations tcal ON tcal.asgloc_id = talc.asgloc_id   
             LEFT JOIN fin_organization_mv tmfom ON tmfom.asgloc_id = cta.call_aloc_id   
             LEFT JOIN asgloc_msn_area_mv tama ON tama.asgloc_id = cta.call_aloc_id   
             LEFT JOIN fin_organization_mv twmfom ON twmfom.orgnbr = tama.wm_orgnbr   
             LEFT JOIN fin_organization_mv tefom ON tefom.orgnbr = tama.eccl_orgnbr  
          ;  
               
          COMMIT;  
               
    /*** store weekly grouped info for PERM serving missionaries ***/      
               
   INSERT INTO msny_weekly_tally   
      ( sweep_dt, created_dt   
      , comp_id, comp_nm  
      , CITIZENSHIP_ID   
      , comp_asgloc_orgnbr, comp_asgloc_orgtype_cd, comp_asgloc_orgtype_nm   
      , comp_asgloc_id, comp_asgloc_nm, comp_asgloc_sens_flg   
      , asgloc_orgnbr, asgloc_id, asgloc_nm   
      , asgloc_area_orgnbr, asgloc_area_nm   
      , asgloc_country_id, asgloc_country_nm, asgloc_sens_flg   
      , within_msn_orgnbr, within_msn_asgloc_id, within_msn_nm, within_msn_sens_flg   
      , eccl_org_orgnbr, eccl_org_nm, eccl_org_sens_flg   
      , msn_currency_used, msn_currency_cd, msn_currency_nm, msn_exchange_rt   
      , msn_local_amt, msn_us_amt, msn_facs_num   
      , pos_spcl_id, prim_pos_nm, prim_pos_pros_only_yn   
      , rspbl_org_nm, rspbl_orgnbr   
      , lang_id, lang_nm   
      , complement_elders, complement_sisters, complement_couples, complement_selders, complement_ssisters, prefield_elders   
      , prefield_sisters, prefield_couples, prefield_selders, prefield_ssisters   
      , mtc_elders, mtc_sisters, mtc_couples, mtc_selders, mtc_ssisters   
      , infield_elders, infield_sisters, infield_couples, infield_selders, infield_ssisters   
      , temp_elders, temp_sisters, temp_couples, temp_selders, temp_ssisters   
      , away_elders, away_sisters, away_couples, away_selders, away_ssisters   
      , lah_couples, lah_selders, lah_ssisters   
      )   
     SELECT * FROM (   
       WITH finsm AS   
      (SELECT * FROM msny_week_tally_gt  WHERE sweep_dt = TRUNC(lv_run_date-7))    
       SELECT    
         sweep_dt, lv_run_date  
       , comp_id, comp_nm  
       , citizenship_id  
        , comp_asgloc_orgnbr, comp_asgloc_orgtype_cd, comp_asgloc_orgtype_nm   
       , comp_asgloc_id, comp_asgloc_nm, comp_asgloc_sens_flg  
       , asgloc_orgnbr , asgloc_id, asgloc_nm    
       , asgloc_area_orgnbr, asgloc_area_nm   
       , asgloc_country_id, asgloc_country_nm, asgloc_sens_flg    
       , within_msn_orgnbr, within_msn_asgloc_id, within_msn_nm, within_msn_sens_flg   
       , eccl_org_orgnbr, eccl_org_nm, eccl_org_sens_flg   
       , msn_currency_used, msn_currency_cd, msn_currency_nm, msn_exchange_rt   
       , msn_local_amt, msn_us_amt, msn_facs_num   
       , pos_spcl_id, prim_pos_nm, prim_pos_pros_only_yn   
       , rspbl_org_nm, rspbl_orgnbr   
       , lang_id, lang_nm   
       , complement_elders, complement_sisters, complement_couples, complement_selders, complement_ssisters  
       , prefield_elders, prefield_sisters, prefield_couples, prefield_selders, prefield_ssisters   
       , mtc_elders, mtc_sisters, mtc_couples, mtc_selders, mtc_ssisters   
       , infield_elders, infield_sisters, infield_couples, infield_selders, infield_ssisters   
       , CASE WHEN misstyp_id = 1 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_elders   
       , CASE WHEN misstyp_id = 2 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_sisters   
       , CASE WHEN misstyp_id = 3 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_couples   
       , CASE WHEN misstyp_id = 4 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_selders   
       , CASE WHEN misstyp_id = 5 THEN NVL(tfsw.msny_count,0) ELSE 0 END AS temp_ssisters   
       , away_elders, away_sisters, away_couples, away_selders, away_ssisters   
       , lah_couples, lah_selders, lah_ssisters   
       FROM     
    ( SELECT    
            fsm.sweep_dt  
          , fsm.comp_id, fsm.comp_nm  
          , fsm.citizenship_id  
          , fsm.asgloc_id  
          , fsm.asgloc_nm    
          , fsm.fund_unit_id,  fsm.comp_asgloc_orgnbr  
          , fsm.comp_asgloc_orgtype_cd  
          , fsm.comp_asgloc_orgtype_nm   
          , fsm.comp_asgloc_sens_flg  
          , fsm.comp_asgloc_id  
          , fsm.comp_asgloc_nm   
          , fsm.asgloc_orgnbr   
          , fsm.asgloc_area_orgnbr  
          , fsm.asgloc_area_nm   
          , fsm.asgloc_country_id  
          , fsm.asgloc_country_nm  
          , fsm.asgloc_sens_flg   
          , fsm.within_msn_orgnbr  
          , fsm.within_msn_asgloc_id  
          , fsm.within_msn_nm  
          , fsm.within_msn_sens_flg   
          , fsm.eccl_org_orgnbr  
          , fsm.eccl_org_nm  
          , fsm.eccl_org_sens_flg   
          , fsm.msn_currency_used  
          , fsm.msn_currency_cd  
          , fsm.msn_currency_nm  
          , fsm.msn_exchange_rt   
          , fsm.msn_local_amt  
          , fsm.msn_us_amt  
          , fsm.msn_facs_num   
          , acpv.posspc_id pos_spcl_id  
          , acpv.primary_pos_name prim_pos_nm  
          , acpv.primary_pos_pros_only prim_pos_pros_only_yn   
          , acpv.responsible_org rspbl_org_nm  
          , acpv.responsible_orgnbr rspbl_orgnbr  
          , acpv.lang_id  
          , acpv.language_name lang_nm    
          , alc.misstyp_id   
          , CASE WHEN alc.misstyp_id = 1 THEN alc.comp_num ELSE 0 END AS complement_elders   
          , CASE WHEN alc.misstyp_id = 2 THEN alc.comp_num ELSE 0 END AS complement_sisters   
          , CASE WHEN alc.misstyp_id = 3 THEN alc.comp_num ELSE 0 END AS complement_couples   
          , CASE WHEN alc.misstyp_id = 4 THEN alc.comp_num ELSE 0 END AS complement_selders   
          , CASE WHEN alc.misstyp_id = 5 THEN alc.comp_num ELSE 0 END AS complement_ssisters   
          , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_elders   
          , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_sisters   
          , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_couples   
          , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_selders   
          , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 11 THEN 1 ELSE 0 END ELSE 0 END) AS prefield_ssisters   
          , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_elders   
          , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_sisters   
          , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_couples   
          , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_selders   
          , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 12 THEN 1 ELSE 0 END ELSE 0 END) AS mtc_ssisters   
          , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_elders   
          , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_sisters   
          , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_couples   
          , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_selders   
          , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NULL OR fsm.temp_comp_id = fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS infield_ssisters   
          , SUM(CASE WHEN alc.misstyp_id = 1 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_elders   
          , SUM(CASE WHEN alc.misstyp_id = 2 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_sisters   
          , SUM(CASE WHEN alc.misstyp_id = 3 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_couples   
          , SUM(CASE WHEN alc.misstyp_id = 4 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_selders   
          , SUM(CASE WHEN alc.misstyp_id = 5 THEN  CASE WHEN fsm.procstat_id = 13 AND (fsm.temp_asgloc_id IS NOT NULL AND fsm.temp_comp_id <> fsm.comp_id) THEN 1 ELSE 0 END ELSE 0 END) AS away_ssisters   
          , SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND alc.misstyp_id = 3 THEN 1 ELSE 0 END) AS lah_couples   
          , SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND alc.misstyp_id = 4 THEN 1 ELSE 0 END) AS lah_selders   
          , SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND alc.misstyp_id = 5 THEN 1 ELSE 0 END) AS lah_ssisters   
          FROM  (SELECT * FROM  msny_week_tally_gt  WHERE sweep_dt =  TRUNC(lv_run_date-7))   fsm   
               JOIN asg_loc_components alc ON alc.comp_id = fsm.comp_id   
               JOIN asgloc_comp_pos_vw acpv ON acpv.comp_id = fsm.comp_id -- join for position info   
          GROUP BY   
            fsm.sweep_dt  
          , fsm.comp_id  
          ,   fsm.comp_nm  
          , fsm.citizenship_id  
          , fsm.asgloc_id  
          , fsm.asgloc_nm  
          , fsm.fund_unit_id  
          , fsm.comp_asgloc_id  
          , fsm.comp_asgloc_nm   
          , fsm.comp_asgloc_orgnbr, fsm.comp_asgloc_orgtype_cd, fsm.comp_asgloc_orgtype_nm   
          , fsm.comp_asgloc_sens_flg   
          , fsm.asgloc_orgnbr   
          , fsm.asgloc_area_orgnbr, fsm.asgloc_area_nm   
          , fsm.asgloc_country_id, fsm.asgloc_country_nm, fsm.asgloc_sens_flg   
          , fsm.within_msn_orgnbr, fsm.within_msn_asgloc_id, fsm.within_msn_nm, fsm.within_msn_sens_flg   
          , fsm.eccl_org_orgnbr, fsm.eccl_org_nm, fsm.eccl_org_sens_flg   
          , fsm.msn_currency_used, fsm.msn_currency_cd, fsm.msn_currency_nm, fsm.msn_exchange_rt   
          , fsm.msn_local_amt, fsm.msn_us_amt, fsm.msn_facs_num   
          , acpv.posspc_id, acpv.primary_pos_name, acpv.primary_pos_pros_only   
          , acpv.responsible_org, acpv.responsible_orgnbr   
          , acpv.lang_id, acpv.language_name   
          , alc.comp_num   
          , alc.misstyp_id  
    )   fsm   
    LEFT OUTER JOIN (SELECT temp_comp_id, COUNT(*) msny_count FROM finsm GROUP BY temp_comp_id) tfsw ON tfsw.temp_comp_id = fsm.comp_id)  ;   
      
    COMMIT;  
      
     /*** store weekly grouped info for TEMP serving missionaries ***/       
       
    INSERT INTO msny_weekly_tally    
      ( sweep_dt, created_dt   
      , comp_id, comp_nm  
      , CITIZENSHIP_ID   
      , comp_asgloc_orgnbr, comp_asgloc_orgtype_cd, comp_asgloc_orgtype_nm   
      , comp_asgloc_id, comp_asgloc_nm, comp_asgloc_sens_flg   
      , asgloc_orgnbr, asgloc_id, asgloc_nm   
      , asgloc_area_orgnbr, asgloc_area_nm   
      , asgloc_country_id, asgloc_country_nm, asgloc_sens_flg   
      , within_msn_orgnbr, within_msn_asgloc_id, within_msn_nm, within_msn_sens_flg   
      , eccl_org_orgnbr, eccl_org_nm, eccl_org_sens_flg   
      , msn_currency_used, msn_currency_cd, msn_currency_nm, msn_exchange_rt   
      , msn_local_amt, msn_us_amt, msn_facs_num   
      , pos_spcl_id, prim_pos_nm, prim_pos_pros_only_yn   
      , rspbl_org_nm, rspbl_orgnbr   
      , lang_id, lang_nm   
      , complement_elders, complement_sisters, complement_couples, complement_selders, complement_ssisters, prefield_elders   
      , prefield_sisters, prefield_couples, prefield_selders, prefield_ssisters   
      , mtc_elders, mtc_sisters, mtc_couples, mtc_selders, mtc_ssisters   
      , infield_elders, infield_sisters, infield_couples, infield_selders, infield_ssisters   
      , temp_elders, temp_sisters, temp_couples, temp_selders, temp_ssisters   
      , away_elders, away_sisters, away_couples, away_selders, away_ssisters   
      , lah_couples, lah_selders, lah_ssisters   
      )   
    SELECT * FROM    
      (   
      WITH ot AS -- "ot" stands for orphaned temporaries   
        (SELECT TRUNC(lv_run_date-7)sw_dt, TRUNC(lv_run_date) cr_dt, fsm.temp_comp_id, fsm.temp_comp_nm,   
                   citizenship_id,  
                   fsm.temp_comp_asgloc_orgnbr, fsm.temp_comp_asgloc_id, fsm.temp_comp_asgloc_nm, fsm.temp_comp_asgloc_sens_flg,   
                   fsm.temp_asgloc_orgnbr, fsm.temp_asgloc_id, fsm.temp_asgloc_nm, fsm.temp_asgloc_sens_flg,   
                   fsm.temp_within_msn_orgnbr, fsm.temp_within_msn_asgloc_id, fsm.temp_within_msn_nm, fsm.temp_within_msn_sens_flg,   
                   fsm.temp_eccl_org_orgnbr, fsm.temp_eccl_org_nm, fsm.temp_eccl_org_sens_flg,   
                   SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND fsm.misstyp_id = 3 THEN 1 ELSE 0 END) AS lah_couples,   
                   SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND fsm.misstyp_id = 4 THEN 1 ELSE 0 END) AS lah_selders,   
                   SUM(CASE WHEN fsm.live_at_home_yn = 'Y' AND fsm.misstyp_id = 5 THEN 1 ELSE 0 END) AS lah_ssisters,   
                   COUNT(*) AS cnt   
              FROM  msny_week_tally_gt  fsm   
             WHERE fsm.sweep_dt = TRUNC(lv_run_date-7)   
               AND fsm.temp_comp_id IS NOT NULL   
               AND NOT EXISTS (SELECT NULL   
                      FROM msny_weekly_tally fst   
                     WHERE fst.sweep_dt = TRUNC(lv_run_date-7)   
                       AND fst.comp_id = fsm.temp_comp_id)   
            GROUP BY    
                   fsm.temp_comp_id, fsm.temp_comp_nm,   
                   citizenship_id,  
                   fsm.temp_comp_asgloc_orgnbr, fsm.temp_comp_asgloc_id, fsm.temp_comp_asgloc_nm, fsm.temp_comp_asgloc_sens_flg,   
                   fsm.temp_asgloc_orgnbr, fsm.temp_asgloc_id, fsm.temp_asgloc_nm, fsm.temp_asgloc_sens_flg,   
                   fsm.temp_within_msn_orgnbr, fsm.temp_within_msn_asgloc_id, fsm.temp_within_msn_nm, fsm.temp_within_msn_sens_flg,   
                   fsm.temp_eccl_org_orgnbr, fsm.temp_eccl_org_nm, fsm.temp_eccl_org_sens_flg   
        )   
      SELECT 
           TRUNC(lv_run_date-7) sw_dt 
         , TRUNC(lv_run_date) cr_dt 
         , ot.temp_comp_id AS comp_id 
         , ot.temp_comp_nm AS comp_nm  
         , citizenship_id   
         , ot.temp_comp_asgloc_orgnbr AS comp_asgloc_orgnbr, cfom.orgtypecode AS comp_asgloc_orgtype_cd, cfom.orgtypeshortdesctxt AS comp_asgloc_orgtype_nm   
         , ot.temp_comp_asgloc_id AS comp_asgloc_id, ot.temp_comp_asgloc_nm  AS comp_asgloc_nm, ot.temp_comp_asgloc_sens_flg AS comp_asgloc_sens_flg   
         , ot.temp_asgloc_orgnbr AS asgloc_orgnbr, ot.temp_asgloc_id AS asgloc_id, ot.temp_asgloc_nm AS asgloc_nm   
         , acom.areaorgnbr AS asgloc_area_orgnbr, acom.areamediumname AS asgloc_area_nm   
         , apc.country_poldivid asgloc_country_id, apc.country_longname AS asgloc_country_nm, ot.temp_asgloc_sens_flg AS asgloc_sens_flg   
         , ot.temp_within_msn_orgnbr AS within_msn_orgnbr, ot.temp_within_msn_asgloc_id AS within_msn_asgloc_id, ot.temp_within_msn_nm AS within_msn_nm, ot.temp_within_msn_sens_flg AS within_msn_sens_flg   
         , ot.temp_eccl_org_orgnbr AS eccl_org_orgnbr, ot.temp_eccl_org_nm AS eccl_org_nm, ot.temp_eccl_org_sens_flg AS eccl_org_sens_flg   
         , afp.cfis_local_or_us_flag AS msn_currency_used, afp.finpol_iso_currency_code AS msn_currency_cd   
         , acurr.currency_desc AS msn_currency_nm, apcr.exch_rate AS msn_exchange_rt   
         , CASE WHEN afp.cfis_local_or_us_flag = 'L' THEN afa.base_amt ELSE ROUND(afa.base_amt / apcr.exch_rate) END AS msn_local_amt   
         , CASE WHEN afp.cfis_local_or_us_flag = 'L' THEN ROUND(afa.base_amt * apcr.exch_rate) ELSE afa.base_amt END AS msn_us_amt   
         , afa.gl_unit_num AS msn_facs_num   
         , acpv.posspc_id pos_spcl_id, acpv.primary_pos_name prim_pos_nm, acpv.primary_pos_pros_only prim_pos_pros_only_yn   
         , acpv.responsible_org rspbl_org_nm, acpv.responsible_orgnbr rspbl_orgnbr   
         , acpv.lang_id, acpv.language_name lang_nm   
         , CASE WHEN alc.misstyp_id = 1 THEN alc.comp_num ELSE 0 END AS complement_elders   
         , CASE WHEN alc.misstyp_id = 2 THEN alc.comp_num ELSE 0 END AS complement_sisters   
         , CASE WHEN alc.misstyp_id = 3 THEN alc.comp_num ELSE 0 END AS complement_couples   
         , CASE WHEN alc.misstyp_id = 4 THEN alc.comp_num ELSE 0 END AS complement_selders   
         , CASE WHEN alc.misstyp_id = 5 THEN alc.comp_num ELSE 0 END AS complement_ssisters   
         -- Missionaries with active temp assignments will always be infield. They can have planned   
         -- temp assignments while prefield or MTC, but those assignments won't show up   
         -- in current temp assignment inline view, which is used in the initial insert to fin_sweep_msny   
         , 0 AS prefield_elders   
         , 0 AS prefield_sisters   
         , 0 AS prefield_couples   
         , 0 AS prefield_selders   
         , 0 AS prefield_ssisters   
         , 0 AS mtc_elders   
         , 0 AS mtc_sisters   
         , 0 AS mtc_couples   
         , 0 AS mtc_selders   
         , 0 AS mtc_ssisters   
         , 0 AS infield_elders   
         , 0 AS infield_sisters   
         , 0 AS infield_couples   
         , 0 AS infield_selders   
         , 0 AS infield_ssisters   
         , CASE WHEN alc.misstyp_id = 1 THEN ot.cnt ELSE 0 END AS temp_elders   
         , CASE WHEN alc.misstyp_id = 2 THEN ot.cnt ELSE 0 END AS temp_sisters   
         , CASE WHEN alc.misstyp_id = 3 THEN ot.cnt ELSE 0 END AS temp_couples   
         , CASE WHEN alc.misstyp_id = 4 THEN ot.cnt ELSE 0 END AS temp_selders   
         , CASE WHEN alc.misstyp_id = 5 THEN ot.cnt ELSE 0 END AS temp_ssisters   
         , 0 AS away_elders   
         , 0 AS away_sisters   
         , 0 AS away_couples   
         , 0 AS away_selders   
         , 0 AS away_ssisters   
         , ot.lah_couples   
         , ot.lah_selders   
         , ot.lah_ssisters   
      FROM ot   
      JOIN assignment_locations al   
           ON al.asgloc_id = ot.temp_asgloc_id   
      JOIN asg_loc_components alc   
           ON alc.comp_id = ot.temp_comp_id   
      JOIN assignment_locations cal   
           ON cal.asgloc_id = alc.asgloc_id   
      JOIN fin_organization_mv cfom -- join for component-level assignment location info   
           ON cfom.orgnbr = cal.asgloc_col_orgnbr   
      JOIN asgloc_comp_pos_vw acpv -- join for position info   
           ON acpv.comp_id = ot.temp_comp_id   
      JOIN fin_organization_mv acom -- join for assignment location info   
           ON acom.orgnbr = al.asgloc_col_orgnbr   
      JOIN poldiv_to_country_mv apc -- join for assignment location country   
           ON apc.primary_poldivid = acom.primarylocpoldivid   
      JOIN fin_poldivs afp -- join for assignment location country financial attributes   
           ON afp.poldivid = apc.country_poldivid   
      JOIN currency_type_vw acurr   
           ON acurr.iso_currency_code = afp.finpol_iso_currency_code   
      JOIN fin_exch_rate apcr   
           ON apcr.from_cur = acurr.iso_currency_code   
      JOIN fin_asglocs afa   
           ON afa.asgloc_id = al.asgloc_id   
      );   
     
   COMMIT;  
    
END weekly_tally;
   
END sweeps;
